

```

1 *****
2 *
3 *           N A D A . P R O D O S
4 *
5 *           ProDOS version of NadaNet
6 *
7 *           Michael J. Mahon - April 14, 1996
8 *           Revised May 3, 2010
9 *
10 * Copyright (c) 1996, 2003, 2004, 2005, 2008, 2009, 2010
11 *
12 * NadaNet is a suite of 6502 machine code routines
13 * to support bidirectional communication among several
14 * Apple // computers. It uses one wire (plus ground)
15 * connected to the game ports of the machines.
16 *
17 * An annunciator output is used to "broadcast" to all
18 * machines, and a "pushbutton" input is used to sense
19 * the state of the shared signalling wire. This is
20 * similar to Ethernet, but at lower speed and at TTL
21 * levels.
22 *
23 * The raw signaling speed is 1 bit every 8 cycles, or
24 * 127.6 kilobaud. With byte separator overhead of 31
25 * cycles, this translates to 1 byte every 94-95 cycles,
26 * or over 10K bytes/sec (thanks to Stephen Thomas!).
27 *
28 * All signal transmission and reception is done with
29 * precisely timed software routines. Synchronization
30 * is assured by a digital PLL at the receiver which
31 * adapts to variations in timing of 93-96 cycles/byte.
32 * (If a machine has a Zip Chip accelerator installed,
33 * it is temporarily slowed during packet transmission
34 * and reception.)
35 *
36 *****
37
38 ***** Version setup *****
39
40 SIZE      equ    $900          ; "Do not exceed" size
41
42          org    $9A00-SIZE ; 'NADANET' for ProDOS hosts
43 master   equ    1            ; Include master-only functions
44 dos      equ    0            ; Non-DOS version
45 crate    equ    0            ; Non-Crate version
46 mserve   equ    0            ; Non-Message Server version
47 ROMboot  equ    0            ; Non-ROM version
48 enhboot  equ    0            ; Non-Enhanced //e ROM version

```

```
50          put    NADAHIST
>1 *****
>2 *
>3 *          Change History
>4 *
>5 *    05/03/10:
>6 *
>7 *    Fixed &TIMEOUT to return to AMPERSAND to set error
>8 *    flags (0) and (1) properly.
>9 *
>10 *    04/28/10:
>11 *
>12 *    Removed version 2.x code from BOOTREQ.
>13 *
>14 *    Added PEEKPOKE request/server for atomic semaphores.
>15 *
>16 *    Changed JSR to service routine in AMPERSAND to go
>17 *    through an indirect JMP to avoid code modification.
>18 *
>19 *    Factored out error retry code for simple requests
>20 *    (SIMPLREQ) to save space in CALLREQ.
>21 *
>22 *    02/10/09 === Released NadaNet 3.0
>23 *
>24 *    01/24/09:
>25 *
>26 *    Added ONERR ($D8) definition and code to clear the
>27 *    flag at boot time and at &RUN time.
>28 *
>29 *    01/06/09:
>30 *
>31 *    Modified NADA.CRATE's NADABOOT2 to coldstart BASIC
>32 *    and set HIMEM to base of NadaNet.
>33 *
>34 *    Changed 'version' to a 2-digit BCD value that is
>35 *    used in messages and the version byte.
>36 *
>37 *    11/11/08:
>38 *
>39 *    Modified RUNSRV to save and restore CSW/KSW hooks
>40 *    so that &RUN works properly with or without an OS.
>41 *
>42 *    11/03/08:
>43 *
>44 *    Added call to FIXLINKS into RUNSRV code to allow
>45 *    BASIC programs to be &RUN at any address > $800.
>46 *
>47 *    10/06/08:
>48 *
>49 *    Added simple BCAST action to SERVER that just sets
>50 *    'address' and 'length' to request values and then
>51 *    returns to the calling code to deal with the data. *
```

```
>52 *
>53 *   Added table of BCAST tags to NADADEFS to serve as a
>54 *   central directory of BCAST tag values.
>55 *
>56 *   09/25/08:
>57 *
>58 *   Added &RUN and &BRUN, as derivatives of &POKE, to
>59 *   run Applesoft programs and M/L programs.
>60 *
>61 *   Added RCVCTL, RCVPTR, rarl=>al, and RCVLONG to the
>62 *   entry point vector for use by BCAST server code.
>63 *
>64 *   09/04/08:
>65 *
>66 *   Restructured SERVER to correct failure to re-sync if
>67 *   'reqctr' was satisfied, and to minimize "deaf" time
>68 *   when iterating in SERVER.
>69 *
>70 *   Added 'reqpidle' (requests per idletime) definition,
>71 *   which is closer to typical.
>72 *
>73 *   08/20/08:
>74 *
>75 *   Added BCAST request as a general mechanism for
>76 *   broadcasting data.
>77 *
>78 *   Added BCASTARB to arbitrate and lock net, then delay
>79 *   for 20ms. to allow all collisions to resolve and any
>80 *   "slow" pollers to get into their RCVCTL holds. This
>81 *   arbitration will precede all broadcast requests, like
>82 *   BOOT, BCAST, and BPOKE.
>83 *
>84 *   08/16/08:
>85 *
>86 *   Changed control packet format:
>87 *   - Combined 'req' and 'mod' bytes into 'rqmd' byte.
>88 *   - Added complement of 'frm', called 'frmc', as a way
>89 *   to detect collisions of synchronized packets.
>90 *   - Removed "delayed BOOTREQ after GETID" boot protocol
>91 *   - Modified BOOTREQ to send old format packet for
>92 *   compatibility with v2.1 PassiveBoot ROM.
>93 *   - Changed sign-on version to v3.0.
>94 *
>95 *   Moved error counters to just after IDTBL, and
>96 *   prefixed them with NadaNet version in hex.
>97 *
>98 *   07/25/08 === Released NadaNet 2.1
>99 *
>100 *   05/21/08:
>101 *
>102 *   Made numerous significant space optimizations:
>103 *   - Added subroutines to set 'address' & 'length'
```

```
>104 *      from most common variables. *
>105 *      - Added PROTERR subroutine to increment count. *
>106 *      - Put checksum counting inside RCVPKT. *
>107 *      - Added variable delay preceding SENDLONG. *
>108 *      - Placed all of the above in "slack" space following *
>109 *      page-aligned SENDPKT and RCVPKT. *
>110 *      - Deleted MONITOR (can always load when needed). *
>111 * *
>112 *      Fixed potential bug if INSTALL called >255 times. *
>113 * *
>114 *      Added ID error checking to 'setid' and INIT. *
>115 * *
>116 *      Changed broadcast BOOTREQ to not be protocol error. *
>117 * *
>118 *      Added 'xmain', 'xsend', and 'xreceive' symbols to *
>119 *      make slack space easy to read in symbol table. *
>120 * *
>121 *      04/21/08: *
>122 * *
>123 *      Added 'svrxkbd' entry to SERVER, used by 'servelp' *
>124 *      to ignore keyboard input. *
>125 * *
>126 *      04/15/08: *
>127 * *
>128 *      Split NADADEFS include file into three parts so that *
>129 *      the control packet definitions could be used without *
>130 *      generating code for the vectors and variables. *
>131 * *
>132 *      Added new "Enhanced boot" protocol for AppleCrate *
>133 *      machines using Enhanced //e's. The new protocol *
>134 *      blindly broadcasts the boot code whenever &BOOTCODE *
>135 *      is invoked. *
>136 * *
>137 *      Since only RCVPKT and RCVLONG, plus RESET and the *
>138 *      actual boot logic need reside in ROM, it fits easily *
>139 *      into the 2 pages of code used by the Enhanced //e's *
>140 *      self-test code. *
>141 * *
>142 *      To support this, a new conditional assembly flag, *
>143 *      'enhboot' has been added and used to select the *
>144 *      code in SENDRCV and NADADEFS for the new boot ROM. *
>145 * *
>146 *      The new AppleCrate boot image will be prefixed with *
>147 *      a second-stage boot that will use GETID to allocate *
>148 *      unique machine IDs. The DACK length hi-byte sent by *
>149 *      Enhanced machines contains a "magic number" ($A5) to *
>150 *      signal that GETID need not schedule a boot code send. *
>151 * *
>152 *      The second stage boot will set up the page 3 RESET *
>153 *      vector to go directly to NadaNet INIT, so it does *
>154 *      not take up space in the running machine. *
>155 * *
```

```
>156 *   The second-stage boot code uses a non-zero 'bootself' *
>157 *   value to indicate that a GETID has already been done *
>158 *   and the second-stage can be skipped for unenhanced *
>159 *   AppleCrate machines. *
>160 * *
>161 *   The maximum number of machines has been increased to *
>162 *   31, and the zeroth entry in the IDTABLE is 31. *
>163 * *
>164 *   07/21/05 === Released NadaNet 2.0 *
>165 * *
>166 *   06/29/05: *
>167 * *
>168 *   Added NadaNet version 2.0 sign-on message to INIT. *
>169 * *
>170 *   Replaced tree-like data send routing in SENDPKT with *
>171 *   new, shorter, lattice-like routine created by Stephen *
>172 *   Thomas. The new routine sends all bits in uniform *
>173 *   cells of 8 cycles, lowering the cycles/byte to 95 *
>174 *   from 106, for a speed increase of more than 11%. *
>175 * *
>176 *   The new data transfer rate is over 10 KB/second. *
>177 * *
>178 *   The packet start synchronization now allows RCVPKT *
>179 *   to establish fine sync for the first byte. *
>180 * *
>181 *   The new RCVPKT samples data bitcells only during the *
>182 *   5th, 6th, and 7th of 8 cycles, making NadaNet much *
>183 *   more tolerant of long network time constants caused *
>184 *   by too much cable or too high a pulldown resistance. *
>185 * *
>186 *   The timing of the check byte is now identical to the *
>187 *   timing of data bytes. *
>188 * *
>189 *   RCVPKT is also changed to reflect the new timings, *
>190 *   which required unrolling the receive code again. *
>191 * *
>192 *   Increased receive-to-send turnaround delays in *
>193 *   PEEKSRV and GETMSRV to allow some margin for the *
>194 *   receiving machine to start polling. *
>195 * *
>196 *   06/08/05: *
>197 * *
>198 *   Fixed bus fight in RCVPKT pointed out by an astute *
>199 *   reader of the code, Stephen Thomas, who also sent *
>200 *   replacement code which only reads the paddle input *
>201 *   and which cleverly combines data shifting with loop *
>202 *   control, permitting the receive code to be re-rolled *
>203 *   to save space! *
>204 * *
>205 *   12/05/04 === Released NadaNet 1.2 *
>206 * *
>207 *   12/01/04: *
```

```
>208 *
>209 * Fixed GETID bug that left 'sbuf+adr' unset if the ID *
>210 * received was not a temporary ID. (For masters only) *
>211 *
>212 * Parameterized maximum number of machines (maxid) and *
>213 * changed GETID so that any ID > maxid is considered a *
>214 * temporary ID to be assigned a permanent ID. *
>215 *
>216 * Changed handling of protocol errors in SERVER so *
>217 * they are "timed" as if they were requests. *
>218 *
>219 * 11/17/04: *
>220 *
>221 * Changed 'servegap' wait time to 3/4 of min arb time *
>222 * to allow some margin for server routine processing *
>223 * after network is released (which is subtracted from *
>224 * "SERVER visible" inter-request gap). *
>225 *
>226 * 11/13/04 *
>227 *
>228 * Changed AmperNada handler to leave return variable *
>229 * unchanged when an error occurs. *
>230 *
>231 * 11/12/04: *
>232 *
>233 * Increased BPOKE locked wait time to 20ms. to allow *
>234 * more "dead time" in an Applesoft &SERVE polling loop.*
>235 *
>236 * 11/10/04: *
>237 *
>238 * Changed SERVER gap wait to wait for an unchanging *
>239 * net, not an idle net, so that a BPOKE is received *
>240 * when preceded by a locked state. *
>241 *
>242 * Fixed bug in PKINCSRV that dropped carry. *
>243 *
>244 * 11/08/04: *
>245 *
>246 * Changed AmperNada handler to throw an Applesoft *
>247 * "DATA" (49) error by default when a command fails. *
>248 * This error can be caught by an active ONERR, or it *
>249 * can be suppressed by appending a "#" to the command. *
>250 * If the error is suppressed, it is the programmer's *
>251 * responsibility to check status by PEEKing 1 and 0. *
>252 *
>253 * 11/06/04: *
>254 *
>255 * Removed &ONERR(err?) because its residual effects-- *
>256 * storing status into variable memory--outlast any *
>257 * running program unless explicitly cancelled. Using *
>258 * PEEK(1) is a safe and effective alternative solution.*
>259 *
```

```
>260 * 11/05/04: *
>261 * *
>262 * Added BPOKE & PEEKINC requestors and servers. *
>263 * *
>264 * Added &IDTBL(val?) to retrieve address of 'idtable' *
>265 * in 'master' version. *
>266 * *
>267 * Changed ARBTRATE to use a single loop, and SETID to *
>268 * use ID for 'arbxv' when a temp ID (>$7F) is used. *
>269 * *
>270 * 11/01/04: *
>271 * *
>272 * Integrated AmperNada ampersand interface for BASIC *
>273 * into NadaNet. Size limit is now $900. *
>274 * *
>275 * 10/27/04: *
>276 * *
>277 * Fixed latent BOOT timing bug in server. *
>278 * *
>279 * Changed SERVER so that it returns after processing *
>280 * any request, in addition to when a key is pressed. *
>281 * *
>282 * Changed SERVER and CALLSRV to do indirect jumps, *
>283 * rather than pushing addresses on stack for rts. *
>284 * *
>285 * Changed REQUEST resend count so that the request *
>286 * timeout set by 'reptime' is accurate. *
>287 * *
>288 * Changed MONITOR to wait for a minimum period of *
>289 * unchanging network state, rather than '0' state, so *
>290 * a locked state between packets is detected as a gap. *
>291 * *
>292 * 10/20/04: *
>293 * *
>294 * Added changes so that NADABOOT could be built using *
>295 * standard NADANET "put" files. *
>296 * *
>297 * Split this change history into a separate file. *
>298 * *
>299 * 10/18/04: *
>300 * *
>301 * Added code to INIT to set up $3CD with warm start *
>302 * 'JMP servelp', so $3CF is NADANET's load page. *
>303 * *
>304 * 10/13/04: *
>305 * *
>306 * Made PUTMREQ and GETMREQ conditional upon 'master' *
>307 * conditional compile flag. PUTMSRV and GETMSRV are *
>308 * conditional upon 'not master'. This frees up space *
>309 * for additional enhancements by splitting NadaNet into *
>310 * different functional subsets for different purposes. *
>311 * *
```

```
>312 *   Made RCVPKT timeout variable so it can be set to the *
>313 *   minimum arbitration time while within a protocol, *
>314 *   to protect the protocol from "outside" interference, *
>315 *   and set to 20ms. outside a protocol, when SERVE or *
>316 *   MONITOR is running, to reduce polling dead time. *
>317 *
>318 *   Moved packet-starting 'ONE' earlier in SENDPKT so *
>319 *   that ARBTRATE and RCVPKT do not need to double-poll *
>320 *   bus to detect start pulse. *
>321 *
>322 *   Changed BOOTREQ to use boot code address, length, *
>323 *   and local address set up prior to SERVER call. *
>324 *
>325 *   Split entry point vector and variable definitions *
>326 *   out into NADADEFS "put" file for use in other progs. *
>327 *
>328 *   10/05/04: *
>329 *
>330 *   Changed ARBTRATE to lock the bus after a successful *
>331 *   poll, so that the arbitration "increment" could be *
>332 *   reduced to 22 cycles from 66. *
>333 *
>334 *   Changed SETID arbitration time calculation to match. *
>335 *
>336 *   Changed SERVER so that received requests, whether *
>337 *   acted upon or not, are counted as 1/8 of a 20ms. *
>338 *   timeout interval. This allows time-related events *
>339 *   to occur properly whether the net is busy or idle. *
>340 *
>341 *   Moved 'sbuf', 'rbuf', and counters so that they will *
>342 *   move less in the future. *
>343 *
>344 *   Made REQUEST retry limit an initialized variable so *
>345 *   that it can be lowered to speed up detection of a *
>346 *   possibly non-existent machine. *
>347 *
>348 *   Moved the 'monch' table used by PUTMSRV and GETMSRV *
>349 *   from internal memory to the unused top of the page *
>350 *   map table, saving 48 bytes of program memory. *
>351 *
>352 *   07/26/04 == Released NadaNet 1.1 *
>353 *
>354 *   06/23/04: *
>355 *
>356 *   Added iteration counter to SERVER and dispensed with *
>357 *   SERVE1. Changed SERVE sync wait to min arbitration *
>358 *   time. *
>359 *
>360 *   Disabled interrupts during SENDPKT and RCVPKT. *
>361 *
>362 *   Made "packet"/"message" nomenclature consistent. *
>363 *
```



```
>364 * 06/04/04: *
>365 * *
>366 * Made INIT, SERVE, and BOOT functions and PEEK, POKE, *
>367 * and CALL functions separate include modules. *
>368 * *
>369 * Added text graphics for protocols. *
>370 * *
>371 * 05/26/04 *
>372 * *
>373 * Added MONITOR function for snooping all packets on *
>374 * the network and logging the first 8 bytes in memory. *
>375 * *
>376 * 05/15/04: *
>377 * *
>378 * Added "master" conditional assembly switch to *
>379 * control newly added boot functions, BOOTREQ and *
>380 * GETIDSRV. *
>381 * *
>382 * 05/06/04: *
>383 * *
>384 * Shortened arbitration time to 1 ms., since almost *
>385 * all protocols have less than 1 ms. delay between *
>386 * packets. Exceptions will "lock" the net by pulling *
>387 * it high until they can respond. This is effectively *
>388 * an extended "start" pulse, and RCVPKT will wait *
>389 * indefinitely for the transition to low. *
>390 * *
>391 * Currently, only PUTMSRV and GETMSRV can take longer *
>392 * than 1 ms. to respond with ACK or NAK, so they must *
>393 * lock the net until their response. *
>394 * *
>395 * The delay from last arbitration poll until beginning *
>396 * of "start" pulse is 54 cycles, so to avoid collision *
>397 * the arbitration delay difference between machines *
>398 * must exceed 54. Since it must be a multiple of 11 *
>399 * cycles, the offset is machine ID * 66 cycles. *
>400 * *
>401 * Since 8-byte data packets are indistinguishable *
>402 * from control packets, and since data packets are *
>403 * never delayed from a preceding control packet by *
>404 * more than 0.5 ms., SERVER must wait for a net "idle" *
>405 * (low) state for 0.5 ms. to ensure that the next pkt *
>406 * it receives is not data. This delay is only needed *
>407 * if the net has not been polled for more than the *
>408 * arbitration delay (~1 ms.). *
>409 * *
>410 * 04/19/04: *
>411 * *
>412 * Changed RCVPKT timeout to 20 ms., since responses *
>413 * are expected in much less. SERVER loop results in *
>414 * "blind" time of less than 0.04 ms. per iteration. *
>415 * *
```



```

51          put      NADACONST
>1      * NadaNet Constant definitions
>2
>3      * Apple ][ definitions
>4
>5      keybd      equ      $C000          ; Keyboard port
>6      kbstrobe  equ      $C010          ; Keyboard strobe
>7      VBL       equ      $C019          ; Vertical blanking
>8      spkr      equ      $C030          ; Speaker toggle
>9      an0       equ      $C058          ; Annunciator 0 base addr
>10     an1       equ      an0+2
>11     an2       equ      an0+4
>12     an3       equ      an0+6
>13     pb0       equ      $C061          ; "Pushbutton" 0 base addr
>14     pb1       equ      pb0+1
>15     pb2       equ      pb0+2
>16     ptrig     equ      $C070          ; Paddle trigger
>17     dsk6off  equ      $C0E8          ; Deselect 5.25" disk in slot 6
>18
>19     * Apple Monitor definitions
>20
>21     CSW       equ      $36            ; Output vector
>22     KSW       equ      $38            ; Input vector
>23     SOFTEV    equ      $3F2          ; Soft re-entry vector
>24     PWREDUP   equ      $3F4          ; Powered-Up check byte
>25
>26     PRBL2     equ      $F94A          ; Display (X) blanks
>27     PREAD     equ      $FB1E          ; Read PDL(X) into Y
>28     HOME      equ      $FC58          ; Clear display
>29     CROUT1    equ      $FD8B          ; Clear to EOL, then CR
>30     PRBYTE    equ      $FDDA          ; Display A as hex byte
>31     COUT      equ      $FDED          ; Display character in A
>32     BELL      equ      $FF3A          ; Beep for 100ms.
>33
>34     * Applesoft definitions
>35
>36     PSTART    equ      $67            ; Start of BASIC prog
>37     VARTAB    equ      $69            ; End prog / start vars
>38     FRETOP    equ      $6F            ; Start of string storage
>39     HIMEM     equ      $73            ; Highest BASIC mem
>40     PROGEND   equ      $AF            ; End of BASIC prog
>41     ONERR     equ      $D8            ; ONERR flag (0 = off)
>42
>43     COLDSTRT  equ      $E000          ; Cold start BASIC
>44     FIXLINKS  equ      $D4F2          ; Fix up BASIC prog links
>45     RUNPROG   equ      $D566          ; RUN Applesoft prog
>46
>47     * Mapping of hardware resources
>48
>49     dsend     equ      an1            ; Data 'send'
>50     drecv     equ      pb1            ; Data 'receive'
>51     zipslow   equ      dsk6off        ; Zip Chip 'slow mode' for 51 ms.

```



```

>53  * Page zero variables
>54
>55  lastidx  equ   $EB           ; Last RCVPKT buffer index
>56  ckbyte   equ   $EC           ; Check byte
>57  ptr       equ   $ED           ; Data buffer pointer (0..leng-1)
>58  address  equ   $FC           ; Scratch addr of local data
>59  length   equ   $FE           ; Scratch length of local data
>60
>61  * Protocol constants
>62
>63  cyperms  equ   1020          ; Cycles per ms. (really 1020.4)
>64
>65  arbtime  equ   1              ; Min arbitration time (ms)
>66  ]cy      equ   arbtime*cyperms ; Arbtime in cycles
>67  ]cpx     equ   11             ; Cycles per X iteration
>68  arbx     equ   ]cy/]cpx      ; X iterations
>69
>70  ]servpad equ   ]cy/4         ; Gap margin
>71  servegap equ   ]cy-]servpad/13 ; SERVER wait loop 13 cyc.
>72
>73  ]cy      equ   ]cpx*256      ; Max arb time (cycles)
>74  maxarb   equ   ]cy+cyperms/cyperms ; ceiling(max arb) (ms)
>75
>76  idletime equ   20            ; Idle polling timeout (ms)
>77                                     ; (stay under 51ms for Zip Chip)
>78  reqdur   equ   6             ; Typical req duration (ms)
>79  reqpidle equ   idletime/reqdur ; Requests per idletime
>80
>81  ]cy      equ   idletime*cyperms ; Timeout in cycles
>82  ]cpx     equ   11            ; Cycles per X iteration
>83  ]cpy     equ   ]cpx*256+4    ; Cycles per Y iteration
>84  idleto   equ   ]cy/]cpy+1    ; Number of Y iterations
>85
>86  reqto    equ   1             ; Timeout within protocol is
>87                                     ; minimum arbitration time.
>88  maxgap   equ   87            ; Max intra-pkt gap (cycles)
>89  gapwait  equ   maxgap/13+1   ; MONITOR wait loop is 13 cyc.
>90
>91  reqtime  equ   3000          ; Req response timeout (ms)
>92  rqperiod equ   20            ; Milliseconds between retries
>93  reqdelay equ   rqperiod-3    ; ARB+SEND+RCV timeout = 3ms.
>94
>95  maxreqrt equ   3             ; Max # of xxxREQ retries
>96  maxretry equ   reqtime/rqperiod/maxreqrt ; # of re-sends

```

```

52          use      NADAMACS
>1          ***** Macro definitions *****
>2
>3          incl6   mac
>4              inc    ]1          ; Increment 16-bit word.
>5              do     ]1+1/$100    ; If ]1 is non-page zero
>6              bne   *+5          ; - No carry.
>7              else   ; Else if ]1 on page zero
>8              bne   *+4          ; - No carry.
>9              fin
>10             inc    ]1+1        ; Propagate carry.
>11             eom
>12
>13          mov16  mac
>14              lda   ]1          ; Move 2 bytes
>15              sta   ]2
>16              if    #=]1
>17              lda   ]1/$100     ; high byte of immediate
>18              else
>19              lda   1+]1
>20              fin
>21              sta   1+]2
>22              eom
>23
>24          delay  mac
>25              ldx   #]1/5       ; (5 cycles per iteration)
>26          ]delay dex
>27              bne   ]delay
>28              eom
>29
>30          dlyms  mac
>31              ldy   #]1         ; Delay 1ms. per iteration
>32          ]dly  delay 1020-4    ; Cycles per ms. - 4
>33              dey
>34              bne   ]dly
>35              eom
>36
>37          align  mac
>38              ds    *-1/]1*]1+]1-*
>39              eom
>40

```

```

53          put    NADADEFS
>1  *****
>2  *
>3  *              NadaNet Definitions
>4  *              v3.1
>5  *
>6  *              Michael J. Mahon - Oct 13, 2004
>7  *              Revised Apr 29, 2010
>8  *
>9  *              Copyright (c) 2004, 2008, 2010
>10 *
>11 *****
>12
>13 version equ    $31          ; NadaNet version 3.1
>14
>15 ***** Control Packet Definition *****
>16
>17          dum    0          ; Control packet format:
0000: 00    >18  rcmd     ds      1          ; Request & Modifier
0001: 00    >19  frmc     ds      1          ; Complement of sending ID
0002: 00    >20  dst      ds      1          ; Destination ID (0 = bcast)
0003: 00    >21  frm      ds      1          ; Sending ID (never 0)
0004: 00 00 >22  adr      ds      2          ; Address field
0006: 00 00 >23  len      ds      2          ; Length field
>24          ; =====
>25  lenctl   ds      0          ; Length of control packet
>26          dend
>27
>28  * Request codes (upper 5 bits) and modifiers (lower 3 bits)
>29
>30  reqfac   equ     8          ; Request code factor (2^3)
>31  reqmask  equ    256-reqfac ; Request code mask (7..3)
>32  modmask  equ    reqfac-1   ; Modifier code mask (2..0)
>33
>34          dum    reqfac      ; Request codes (0 invalid):
0008: 00 00 00 >35  r_PEEK   ds      reqfac      ; PEEK request
0010: 00 00 00 >36  r_POKE   ds      reqfac      ; POKE request
0018: 00 00 00 >37  r_CALL   ds      reqfac      ; CALL request
0020: 00 00 00 >38  r_PUTMSG ds      reqfac      ; PUTMSG request
0028: 00 00 00 >39  r_GETMSG ds      reqfac      ; GETMSG request
0030: 00 00 00 >40  r_GETID  ds      reqfac      ; GETID request
0038: 00 00 00 >41  r_BOOT   ds      reqfac      ; BOOT request (in ROM)
0040: 00 00 00 >42  r_BCAST  ds      reqfac      ; BCAST request
0048: 00 00 00 >43  r_BPOKE  ds      reqfac      ; Broadcast POKE request
0050: 00 00 00 >44  r_PKINC  ds      reqfac      ; PEEK & INCrement request
0058: 00 00 00 >45  r_PKPOK  ds      reqfac      ; PEEKPOKE request
0060: 00 00 00 >46  r_RUN    ds      reqfac      ; RUN request
0068: 00 00 00 >47  r_BRUN   ds      reqfac      ; BRUN request
>48          ; =====
>49  maxreq   ds      0          ; Max request + reqfac
>50          dend
>51

```

```

>52          dum    1          ; Modifier codes (0 invalid):
0001: 00    >53  rm_REQ   ds    1          ; Request
0002: 00    >54  rm_ACK   ds    1          ; Acknowledge
0003: 00    >55  rm_DACK  ds    1          ; Data Acknowledge
0004: 00    >56  rm_NAK   ds    1          ; Negative Acknowledge
>57          dend
>58
>59          ***** BCAST tags *****
>60          *
>61          * High byte of BCAST address field.  Tags <$D0 *
>62          * can be confused with RAM addresses. (The low *
>63          * byte may be an additional specification.) *
>64          *
>65          *****
>66
>67  t_BASIC  equ    $E0          ; Applesoft BASIC program
>68  t_SYNTH  equ    $F0          ; Crate SYNTH program
>69  t_VOICE  equ    $F1          ; Crate SYNTH voice
>70
>71          ***** NadaNet Page 3 Vector *****
>72
>73          dum    $3CC          ; Fixed memory vector
03CC: 00    >74  bootself db    0          ; Machine ID from BOOT
03CD: 4C 00 00 >75  warmstrt jmp    0*0          ; Warm start SERVE loop entry
>76  nadapage equ    *-1          ; NADANET load page
>77          dend

```



```

    54          put    NADAVECTOR
    >1          ***** Entry Points *****
    >2
9100: 20 29 92 >4  entry   jsr    INSTALL   ; BOOT entry: init and
9103: 20 03 94 >5  servelp  jsr    svrxkbd   ; SERVE ignoring keypresses
9106: 4C 03 91 >6          jmp    servelp   ; forever...
    >7
9109: 4C 29 92 >8  init     jmp    INSTALL   ; Initialize and return
910C: 4C 06 94 >9  serve    jmp    SERVER    ; Run request server
910F: 4C E3 94 >11  peek     jmp    PEEKREQ    ;
9112: 4C 7F 95 >12  poke     jmp    POKEREQ    ;
9115: 4C 31 96 >13  call     jmp    CALLREQ   ;
9118: 4C C7 96 >14  putmsg   jmp    PUTMREQ    ;
911B: 4C E9 96 >15  getmsg   jmp    GETMREQ    ;
911E: 4C A1 94 >16  bcast    jmp    BCASTREQ   ;
9121: 4C 17 96 >17  bpoke    jmp    BPOKEREQ   ;
9124: 4C 37 95 >18  peekinc  jmp    PKINCREQ   ;
9127: 4C 3B 95 >19  peekpoke jmp    PKPOKREQ   ;
912A: 4C 77 95 >20  run      jmp    RUNREQ     ;
912D: 4C 7B 95 >21  brun     jmp    BRUNREQ    ;
9130: 4C 00 99 >22  rcvctl   jmp    RCVCTL     ;
9133: 4C 0A 99 >23  rcvptr   jmp    RCVPTR     ;
9136: 4C 98 99 >24  RARL=>AL jmp    rarl=>al   ;
9139: 4C CF 99 >25  rcvlong  jmp    RCVLONG    ;
    >42
    55          put    NADAVARS
    >1          ***** Parameters and variables *****
    >2
    >6
913C: 00          >7  self     db      0          ; Our own machine ID
913D: 00 00 00 >8  sbuf     ds     lenctl    ; Control pkt send buffer
9145: 00 00 00 >9  rbuf     ds     lenctl    ; Control pkt receive buffer
914D: 00 00          >10  locaddr  dw     0          ; Local address of req data
914F: 32          >11  retrylim db     maxretry   ; Limit of REQUEST resends
9150: 00          >12  servecnt db     0          ; SERVE iterations (0=256)
    >13
    >14  parmsiz equ    *-self   ; Size of parameter area
    >15
    >16          ***** Counters and Version *****
    >17
9151: 5C          >18  arbxv    db     arbx      ; Arbitrate X iters (modified)
9152: 01          >19  tolim    db     reqto    ; RCVPKT timeout limit
9153: 03          >20  reqctr   db     reqpidle  ; SERVER request counter
9154: 00          >21  reqretry db     0          ; xxxREQ retries remaining
9155: 00          >22  retrycnt db     0          ; REQUEST resend count
9156: 00 00        >23  errprot  dw     0          ; Protocol error count
9158: 00 00        >24  ckerr    dw     0          ; Checksum error count
915A: 00 00        >25  frmcerr  dw     0          ; 'frmc' collision errors
915C: 31          >26  nadaver  db     version   ; NadaNet version
    >27
    >29          * Table of allocated machine IDs (allocated = non-zero)
    >30

```

```
>31 maxid equ 31 ; Maximum number of machines
>32
915D: 1F 04 >33 idtable db maxid,4+dos ; Table of machine attributes
915F: 00 00 00 >34 ds maxid-1 ; Rest of ID table (=0)
>39
56 put AMPERSAND
```

```

>2 *****
>3 *
>4 *           A M P E R N A D A
>5 *
>6 *           Michael J. Mahon - Oct 25, 2004
>7 *           Revised May 3, 2010
>8 *
>9 *           Copyright (c) 2004, 2008, 2010
>10 *
>11 * Implements an ampersand (&) interface to NadaNet for
>12 * Applesoft programs. Reduces the need for PEEKs and
>13 * POKEs to set up parameters, saving time and interface
>14 * definitions.
>15 *
>16 * If an error occurs in a command execution routine,
>17 * (signaled by Carry set upon return) the handler will,
>18 * by default, throw a "DATA" (49) error, which will halt
>19 * the program unless caught by an active ONERR.
>20 *
>21 * If an ampersand command is followed by a "#", then no
>22 * execution error will be thrown, and the programmer
>23 * is responsible for checking status by PEEKing 1 and 0.
>24 *
>25 *****
>26
>27 ***** Applesoft Definitions *****
>28
>29 TXTPTR equ $B8 ; Current scan point
>30 VALTYP equ $11 ; $FF if var is STRING$
>31 INTFLG equ $12 ; $80 if var is INT%
>32 FORPNT equ $85 ; Ptr to var
>33 FAC equ $9D ; Floating point accum
>34
>35 AMPVECT equ $3F5 ; JMP to ampersand handler
>36
>37 CHRGET equ $00B1 ; Get next text char
>38 CHRGOT equ $00B7 ; Get last text char
>39 ERROR equ $D412 ; Applesoft error handler
>40 SYNERR equ $DEC9 ; Syntax Error
>41 ADDON equ $D998 ; Advance TXTPTR by Y
>42 SYNCHR equ $DEC0 ; Current char must = A
>43 FRMNUM equ $DD67 ; Eval expr to FAC
>44 PTRGET equ $DFE3 ; Get var, ptr in (Y,A)
>45 GETBYT equ $E6F8 ; Eval expr to X
>46 GETADR equ $E752 ; Eval expr to (Y,A)
>47 FLO2 equ $EBA0 ; Normalize FAC (C set)
>48 SETFOR equ $EB27 ; Pack FAC to (FORPNT)
>49
>50 ***** Variables *****
>51
>52 cmdptr equ $EC ; Cmd table cursor
>53 cmdsave equ $ED ; Current parm descriptor

```

```
>54  disp      equ    $EF      ; Displacement to parm value
>55
917D: 00      >56  instald  db     0      ; Installed flag
917E: 00      >57  nparms   db     0      ; # of parms seen
917F: 00      >58  errstop  db     0      ; "Throw error" flag
9180: 00      >59  varcmd   db     0      ; var parm descriptor
9181: 00      >60  vartype  db     0      ; variable type
9182: 00 00   >61  varadr   da     0      ; variable address
```

```

>63 ***** Ampersand Command Table *****
>64
>65 * Applesoft Token Definitions
>66
>67 CALL_t    equ    140
>68 RUN_t     equ    172
>69 POKE_t    equ    185
>70 GET_t     equ    190
>71 PEEK_t    equ    226
>72
>73 * Syntax string definitions
>74
>75 @         equ    self-1      ; NadaNet parameter origin
>76 byte     equ    $00         ; Byte
>77 word     equ    $40         ; Word
>78 var      equ    $80         ; Numeric variable
>79
>80         err    parmsiz/63 ; Parm area < 64 bytes
>81
>82 iter     equ    servecnt-@.byte ; SERVER iteration count
>83 dest     equ    sbuf+dst-@.byte ; Destination machine
>84 addr     equ    sbuf+adr-@.word ; Address at destination
>85 lngth    equ    sbuf+len-@.word ; Length
>86 locadr   equ    locaddr-@.word ; Local address
>87 AX       equ    sbuf+len-@.word ; A,X regs for CALL
>88 class    equ    sbuf+adr-@.word ; Class of message
>89 incr     equ    sbuf+len-@.word ; Increment for PEEK INC
>90 val      equ    sbuf+len-@.word ; Value for BPOKE, PEEKPOKE
>91 n60ms    equ    retrylim-@.byte ; Request resend limit
>92 lngth?   equ    rbuf+len-@.word.var ; Length (var)
>93 val?     equ    rbuf+len-@.word.var ; Value (var)
>94
>95 * In command table, longer commands must precede shorter
>96 * commands with a common prefix.
>97
9184: 53 45 52 >98 cmdtable asc  'SERVE',00          ; &SERVE
918A: 15 00    >99         db    iter,0
918C: 06 94    >100        da    SERVER
>101
918E: 50 55 54 >102        asc  'PUTMSG',00          ; &PUTMSG
9195: 04 46 48 >103        db    dest,class,lngth,locadr,0
919A: C7 96    >104        da    PUTMREQ
>105
919C: BE 4D 53 >106        db    GET_t,'M','S','G',0      ; &GETMSG
91A1: 04 46 D0 >107        db    dest,class,lngth?,locadr,0
91A6: E9 96    >108        da    GETMREQ
>109
91A8: E2 49 4E >110        db    PEEK_t,'I','N','C',0      ; &PEEKINC
91AD: 04 46 48 >111        db    dest,addr,incr,val?,0
91B2: 37 95    >112        da    PKINCREQ
>113
91B4: E2 B9 00 >114        db    PEEK_t,POKE_t,0          ; &PEEKPOKE

```

```

91B7: 04 46 48 >115      db      dest,addr,val,val?,0
91BC: 3B 95      >116      da      PKPOKREQ
          >117
91BE: E2 00      >118      db      PEEK_t,0          ; &PEEK
91C0: 04 46 48 >119      db      dest,addr,length,locadr,0
91C5: E3 94      >120      da      PEEKREQ
          >121
91C7: B9 00      >122      db      POKE_t,0          ; &POKE
91C9: 04 46 48 >123      db      dest,addr,length,locadr,0
91CE: 7F 95      >124      da      POKEREQ
          >125
91D0: AC 00      >126      db      RUN_t,0          ; &RUN
91D2: 04 46 48 >127      db      dest,addr,length,locadr,0
91D7: 77 95      >128      da      RUNREQ
          >129
91D9: 42 AC 00 >130      db      'B',RUN_t,0      ; &BRUN
91DC: 04 46 48 >131      db      dest,addr,length,locadr,0
91E1: 7B 95      >132      da      BRUNREQ
          >133
91E3: 8C 00      >134      db      CALL_t,0          ; &CALL
91E5: 04 46 48 >135      db      dest,addr,AX,0
91E9: 31 96      >136      da      CALLREQ
          >137
91EB: 42 4F 4F >138      asc     'BOOT',00        ; &BOOT
91F0: 46 48 52 >139      db      addr,length,locadr,0
91F4: 9D 94      >140      da      BOOTREQ
          >141
91F6: 42 43 41 >142      asc     'BCAST',00       ; &BCAST
91FC: 46 48 52 >143      db      addr,length,locadr,0
9200: A1 94      >144      da      BCASTREQ
          >145
9202: 42 B9 00 >146      db      'B',POKE_t,0     ; &BPOKE
9205: 46 48 00 >147      db      addr,val,0
9208: 17 96      >148      da      BPOKEREQ
          >149
920A: 49 4E 49 >150      asc     'INIT',00        ; &INIT
920F: 00          >151      db      0
9210: B7 93      >152      da      INIT
          >153
9212: 54 49 4D >154      asc     'TIMEOUT',00     ; &TIMEOUT
921A: 14 00      >155      db      n60ms,0
921C: 72 93      >156      da      timeout
          >157
921E: 49 44 54 >158      asc     'IDTBL',00       ; &IDTBL
9224: D0 00      >159      db      val?,0
9226: 7E 93      >160      da      idtbl
          >161
9228: 00          >162      db      0                ; End of Command Table

```

```

>164 *****
>165 *
>166 *           I N S T A L L
>167 *
>168 *           Michael J. Mahon - Oct 25, 2004
>169 *           Revised Aug 16, 2008
>170 *
>171 *           Copyright (c) 2004, 2008
>172 *
>173 *   Installs AmperNada as first ampersand routine (if not
>174 *   installed already) and chains to an existing routine.
>175 *   if no routine is currently installed, it defaults to
>176 *   "SYNTAX ERROR".
>177 *
>178 *****
>179

```

```

9229: AD 7D 91 >180  INSTALL  lda   instald   ; AmperNada installed?
922C: D0 23   >181          bne   :exit     ; -Yes, don't repeat.
922E: A9 4C   >182          lda   #$4C      ; -No, set flag and install.
9230: 8D 7D 91 >183          sta   instald
9233: CD F5 03 >184          cmp   AMPVECT   ; Is "&" vector a JMP?
9236: 8D F5 03 >185          sta   AMPVECT   ; (always set "jmp")
9239: D0 0C   >186          bne   :setvect  ; -No, just set vector.
          >187  :chain  movl6 AMPVECT+1;chain+1 ; -Yes, chain to it.
923B: AD F6 03 >187          lda   AMPVECT+1 ; Move 2 bytes
923E: 8D 6B 92 >187          sta   chain+1
9241: AD F7 03 >187          lda   1+AMPVECT+1
9244: 8D 6C 92 >187          sta   1+chain+1
          >187          eom
          >188  :setvect movl6 #AMPNADA;AMPVECT+1 ; set the vector.
9247: A9 54   >188          lda   #AMPNADA  ; Move 2 bytes
9249: 8D F6 03 >188          sta   AMPVECT+1
924C: A9 92   >188          lda   #AMPNADA/$100 ; high byte of immediate
924E: 8D F7 03 >188          sta   1+AMPVECT+1
          >188          eom
9251: 4C B7 93 >189  :exit   jmp   INIT      ; Initialize NadaNet.

```

```

>191 *****
>192 *
>193 *           A M P E R N A D A
>194 *
>195 *           Michael J. Mahon - Oct 25, 2004
>196 *           Revised Nov 08, 2004
>197 *
>198 *           Copyright (c) 2004
>199 *
>200 * Implements an ampersand (&) interface to NadaNet for
>201 * Applesoft programs. Reduces the need for PEEKs and
>202 * POKEs to set up parameters, saving time and interface
>203 * definitions.
>204 *
>205 *****
>206

```

```

9254: 08      >207  AMPNADA  php           ; Save status
9255: 48      >208           pha           ; and A for chain.
9256: A2 00   >209           ldx          #0
9258: 8E 7E 91 >210           stx          nparms      ; # of parms supplied
925B: 8E 80 91 >211           stx          varcmd     ; Signal no var params seen
925E: 8E 7F 91 >212           stx          errstop    ; Clear "throw err" flag.
9261: A0 00   >213  cmd      ldy          #0           ; Start compare at TXTPTR
9263: BD 84 91 >214           lda          cmdtable,x ; Get command char
9266: D0 05   >215           bne          comp       ; -Not end, compare.
9268: 68      >216           pla           ; -End. Restore A
9269: 28      >217           plp           ; and status and chain
926A: 4C C9 DE >218  chain    jmp          SYNERR      ; to next & handler.
>219
926D: D1 B8   >220  comp      cmp          (TXTPTR),y ; Does cmd match text?
926F: D0 09   >221           bne          :skipcmd   ; -No, skip this one.
9271: C8      >222           iny           ; -Yes, advance.
9272: E8      >223           inx           ;
9273: BD 84 91 >224           lda          cmdtable,x ; End of command?
9276: D0 F5   >225           bne          comp       ; -No, keep comparing.
9278: F0 11   >226           beq          :doit      ; -Yes, go do it.
>227
927A: E8      >228  :skipcmd  inx           ; Skip to end of
927B: BD 84 91 >229           lda          cmdtable,x ; current cmd string
927E: D0 FA   >230           bne          :skipcmd
9280: E8      >231  :skipp    inx           ; Skip to end of
9281: BD 84 91 >232           lda          cmdtable,x ; current parm vect
9284: D0 FA   >233           bne          :skipp
9286: E8      >234           inx           ; Pass end mark
9287: E8      >235           inx           ; and action
9288: E8      >236           inx           ; routine address.
9289: D0 D6   >237           bne          cmd        ; Go check next command.
>238
928B: 68      >239  :doit     pla           ; Discard entry A
928C: 68      >240           pla           ; and status.
928D: B1 B8   >241           lda          (TXTPTR),y ; Look at next character.
928F: C8      >242           iny           ; (provisional match)

```



```

9290: C9 23      >243      cmp      #'#'      ; Is it "#"?
9292: F0 04      >244      beq      :advance  ; -Yes, don't throw error.
9294: 88         >245      dey      ; -No, don't match, and
9295: EE 7F 91   >246      inc      errstop   ; set throw err flag.
9298: 20 98 D9   >247      :advance jsr      ADDON   ; Advance TXTPTR past cmd
929B: A9 28      >248      lda      #'('      ; Require initial "("
929D: 20 C0 DE   >249      :nxparm  jsr      SYNCHR   ; Syntax err if no match.
92A0: F0 61      >250      beq      :synerr   ; End not expected.
92A2: 86 EC      >251      stx      cmdptr    ; Save for :done case
92A4: C9 29      >252      cmp      #'')'     ; Found a ")"?
92A6: F0 5E      >253      beq      :done     ; -Yes, end of parm list.
92A8: EE 7E 91   >254      inc      nparms    ; -No, another parm.
92AB: E8         >255      inx      ; Advance ptr and
92AC: BD 84 91   >256      lda      cmdtable,x ; get parm descriptor.
92AF: F0 52      >257      beq      :synerr   ; Too many parms.
92B1: 85 ED      >258      sta      cmdsave   ; Save descriptor
92B3: 29 3F      >259      and      #$3F      ; Mask displacement
92B5: 85 EF      >260      sta      disp      ; and save it.
92B7: 86 EC      >261      stx      cmdptr    ; Save pointer.
92B9: 24 ED      >262      bit      cmdsave   ; Test parm type.
92BB: 30 20      >263      bmi      :var      ; -Var parm
92BD: 50 12      >264      bvc      :byte     ; -Byte value parm
92BF: 20 67 DD   >265      jsr      FRMNUM    ; -Word value parm
92C2: 20 52 E7   >266      jsr      GETADR    ; Word val to Y,A
92C5: A6 EF      >267      ldx      disp      ;
92C7: 9D 3C 91   >268      sta      @+1,x     ; Store the value
92CA: 98         >269      tya      ;
92CB: 9D 3B 91   >270      sta      @,x       ;
92CE: 4C F4 92   >271      jmp      :more?    ;
          >272
92D1: 20 F8 E6   >273      :byte  jsr      GETBYT ; Byte value to X
92D4: A4 EF      >274      ldy      disp      ;
92D6: 8A         >275      txa      ;
92D7: 99 3B 91   >276      sta      @,y       ; Store the value
92DA: 4C F4 92   >277      jmp      :more?    ;
          >278
92DD: A5 ED      >279      :var   lda      cmdsave   ; Save the parm
92DF: 8D 80 91   >280      sta      varcmd    ; descriptor.
92E2: 20 E3 DF   >281      jsr      PTRGET    ; Get var ptr in (A,Y)
92E5: 8D 82 91   >282      sta      varadr    ; and save var
92E8: 8C 83 91   >283      sty      varadr+1  ; address.
92EB: A5 11      >284      lda      VALTYP    ; $FF if string
92ED: D0 14      >285      bne      :synerr   ; String not allowed.
92EF: A5 12      >286      lda      INTFLG    ; $80 if INT%
92F1: 8D 81 91   >287      sta      vartype   ; Save for later use
92F4: 20 B7 00   >288      :more? jsr      CHRGOT  ; Check current test char.
92F7: F0 0A      >289      beq      :synerr   ; End not expected.
92F9: C9 29      >290      cmp      #'')'     ; Closing ")"?
92FB: F0 09      >291      beq      :done     ; -Yes, finish.
92FD: A6 EC      >292      ldx      cmdptr    ; -No, more parms.
92FF: A9 2C      >293      lda      #','      ; Require a comma.
9301: D0 9A      >294      bne      :nxparm   ; (always)

```

```

>295
9303: 4C C9 DE >296 :synerr jmp SYNERR ; SYNTAX ERROR
>297
9306: 20 B1 00 >298 :done jsr CHRGET ; Pass the ")"
9309: A6 EC >299 ldx cmdptr
930B: E8 >300 :skipit inx ; Skip to end
930C: BD 84 91 >301 lda cmdtable,x ; of parm descriptors.
930F: D0 FA >302 bne :skipit
>303 movl6 cmdtable+1,x;$00 ; Action routine
9311: BD 85 91 >303 lda cmdtable+1,x ; Move 2 bytes
9314: 85 00 >303 sta $00
9316: BD 86 91 >303 lda 1+cmdtable+1,x
9319: 85 01 >303 sta 1+$00
>303 eom
931B: 20 31 93 >304 jsr :jmp ; Call the action routine
931E: 85 00 >305 sta $00 ; Save returned A
9320: A9 00 >306 lda #0
9322: 2A >307 rol ; C to low bit
9323: 85 01 >308 sta $01 ; Save returned Carry
9325: F0 0D >309 beq :noerr ; No error, continue.
9327: AD 7F 91 >310 lda errstop ; Throw error?
932A: F0 0D >311 beq :rts ; -No, just return.
932C: A2 31 >312 ldx #49 ; -Yes, throw "DATA"
932E: 4C 12 D4 >313 jmp ERROR ; error.
>314
9331: 6C 00 00 >315 :jmp jmp ($00) ; To action routine.
>316
9334: AD 80 91 >317 :noerr lda varcmd ; Var parm passed?
9337: D0 01 >318 bne :store ; -Yes, store into it.
9339: 60 >319 :rts rts ; -No, return.
>320
933A: 29 3F >321 :store and #$3F ; Mask displacement
933C: A8 >322 tay
933D: B9 3B 91 >323 lda @,y ; Get low byte
9340: AA >324 tax ; X = lo byte of value
9341: A9 00 >325 lda #0 ; Hi byte if byte value
9343: 2C 80 91 >326 bit varcmd ; Is it byte or word?
9346: 50 03 >327 bvc :byteval ; -Byte, use 0 hi byte
9348: B9 3C 91 >328 lda @+1,y ; -Word, get hi byte
934B: A8 >329 :byteval tay ; Y = hi byte of value
>330 movl6 varadr;FORPNT ; Address of variable
934C: AD 82 91 >330 lda varadr ; Move 2 bytes
934F: 85 85 >330 sta FORPNT
9351: AD 83 91 >330 lda 1+varadr
9354: 85 86 >330 sta 1+FORPNT
>330 eom
9356: AD 81 91 >331 lda vartype ; INT% or FLOAT variable?
9359: 10 0A >332 bpl :float ; -FLOAT
935B: 98 >333 tya ; -INT%
935C: A0 00 >334 ldy #0 ; Store hi byte
935E: 91 85 >335 sta (FORPNT),y ; in INT% variable.
9360: C8 >336 iny ; Point to lo byte

```

```
9361: 8A      >337      txa                ; Store lo byte
9362: 91 85   >338      sta (FORPNT),y    ; in INT% variable.
9364: 60      >339      rts
          >340
9365: 84 9E   >341 :float  sty FAC+1        ; Hi byte to FAC
9367: 86 9F   >342      stx FAC+2        ; Lo byte to FAC
9369: A2 90   >343      ldx #$90         ; Binary point 16 bits right
936B: 38      >344      sec              ; (Don't negate FAC)
936C: 20 A0 EB >345      jsr FLO2         ; Normalize FAC
936F: 4C 27 EB >346      jmp SETFOR       ; Pack FAC into variable.
```

```
>348 *****
>349 *
>350 *           &TIMEOUT ([n60ms])
>351 *
>352 *           Michael J. Mahon - Oct 28, 2004
>353 *           Revised May 3, 2010
>354 *
>355 *           Copyright (c) 2004, 2010
>356 *
>357 *   Set new request timeout value in units of 60 ms.
>358 *
>359 *   If no value is supplied, reset timeout to default.
>360 *
>361 *****
>362
9372: AD 7E 91 >363 timeout lda  nparms      ; Parm supplied?
9375: D0 05    >364         bne  null       ; -Yes, timeout set.
9377: A9 32    >365         lda  #maxretry ; -No, restore
9379: 8D 4F 91 >366         sta  retrylim  ; the default.
937C: 18      >367 null    clc           ; No error
937D: 60      >368         rts
```

```
>370 *****
>371 *
>372 *          &IDTBL (val?)
>373 *
>374 *          Michael J. Mahon - Nov 05, 2004
>375 *
>376 *          Copyright (c) 2004
>377 *
>378 * Return address of 'idtable' in parm variable.
>379 *
>380 *****
```

```
>381
>382 idtbl    movl6 #idtable;rbuf+len ; Put addr in rbuf
937E: A9 5D >382    lda    #idtable ; Move 2 bytes
9380: 8D 4B 91 >382    sta    rbuf+len
9383: A9 91 >382    lda    #idtable/$100 ; high byte of immediate
9385: 8D 4C 91 >382    sta    1+rbuf+len
>382    eom
9388: 18 >383    clc
9389: 60 >384    rts
```

```

    57          put    INITSERVE
>2    *** Table of service routines used by SERVER ***
>3
938A: 14 95  >4    service dw    PEEKSRV    ; Table of service routines
938C: BC 95  >5          dw    POKE SRV    ; (Must be in order)
938E: 36 96  >6          dw    CALLSRV
9390: 71 94  >11         dw    ]PROTERR    ; (Error if PUTMSG)
9392: 71 94  >12         dw    ]PROTERR    ; (Error if GETMSG)
9394: B4 94  >15         dw    GETIDSRV    ; Master serves GETID
9396: 71 94  >19         dw    ]PROTERR    ; (Error if non-bcast BOOT)
9398: 98 99  >20         dw    rarl=>al    ; (BCAST data up to caller)
939A: 22 96  >21         dw    BPOKESRV
939C: 54 95  >22         dw    PKINCSRV
939E: 58 95  >23         dw    PKPOKSRV
93A0: A7 95  >27         dw    RUNSRV
93A2: BC 95  >29         dw    BRUNSRV
>30
>31    * Version message printed by INIT
>32
93A4: CE C1 C4 >33    vermsg  asc    "NADANET "
93AC: B3      >34          db    version/16."0" ; Major version #
93AD: AE      >35          asc    "."
93AE: B1      >36          db    version&$0F."0" ; Minor version #
93AF: AC A0 C9 >40         asc    ", ID = $"
>41    verlen  equ    *-vermsg    ; Length of msg

```

```

>43 *****
>44 *
>45 *           I N I T
>46 *
>47 *           Michael J. Mahon - Mar 5, 2004
>48 *           Revised May 21, 2008
>49 *
>50 *           Copyright (c) 1996, 2004, 2005, 2008
>51 *
>52 * Initialize NADANET, sign on, and return to caller.
>53 *
>54 *****
>55

```

```

93B7: AD CC 03 >56  INIT      lda    bootself    ; Set up ID from BOOT
93BA: 20 E9 93 >57          jsr    setid
93BD: B0 29      >58          bcs    :err          ; Bad ID, no INIT.
93BF: A9 4C      >59          lda    #$4C          ; Set warmstrt JMP to
93C1: 8D CD 03 >60          sta    warmstrt     ; servlp (& nadapage)
>61          movl6 #servelp;warmstrt+1
93C4: A9 03      >61          lda    #servelp     ; Move 2 bytes
93C6: 8D CE 03 >61          sta    warmstrt+1
93C9: A9 91      >61          lda    #servelp/$100 ; high byte of immediate
93CB: 8D CF 03 >61          sta    1+warmstrt+1
>61          eom
93CE: 20 8B FD >62          jsr    CROUT1      ; New line.
93D1: A0 00      >63          ldy    #0
93D3: B9 A4 93 >64  :msgloop  lda    vermsg,y    ; Print version message
93D6: 20 ED FD >65          jsr    COUT
93D9: C8          >66          iny
93DA: C0 13      >67          cpy    #verlen
93DC: 90 F5      >68          bcc    :msgloop
93DE: AD 3C 91 >69          lda    self        ; and current ID.
93E1: 20 DA FD >70          jsr    PRBYTE      ; (in hex)
93E4: 20 8B FD >71          jsr    CROUT1      ; New line.
93E7: 18          >72          clc                ; Good return.
93E8: 60          >73  :err      rts

```

```

>76 *****
>77 *
>78 *           S E T I D
>79 *
>80 *           Michael J. Mahon - May 13, 2004
>81 *           Revised Aug 17, 2008
>82 *
>83 *           Copyright (c) 2004, 2008
>84 *
>85 * Set machine ID to contents of A register and reset
>86 * the arbitration delay to 'arbtime' plus 22 cycles
>87 * times the machine ID, to avoid collisions.
>88 *
>89 * Delay from last arbitration poll to bus lock is 10
>90 * cycles, so 22 (2 * 11 cycles) increment provides a
>91 * little insurance.
>92 *
>93 *****
>94

```

```

93E9: 8D 3C 91 >95  setid  sta  self      ; Machine ID
93EC: 8D 40 91 >96      sta  sbuf+frm  ; Set sender field.
93EF: 49 FF      >97      eor  #$FF     ; Complement ID
93F1: 8D 3E 91 >98      sta  sbuf+frmc ; for collision detect.
93F4: 49 FF      >99      eor  #$FF     ; Back to ID
93F6: 38          >100     sec          ; Anticipate error.
93F7: F0 09      >101     beq  :err     ; -Error if zero.
93F9: 18          >102     clc          ; Anticipate no error.
93FA: 30 03      >103     bmi  :setarb  ; -Use temp ID (>127)
93FC: 0A          >104     asl          ; Mult ID by 2
93FD: 69 5C      >105     adc  #arbx    ; and add to base
93FF: 8D 51 91 >106     :setarb sta  arbxv  ; arb delay.
9402: 60          >107     :err  rts

```



```

>110 *****
>111 *
>112 *           S E R V E R
>113 *
>114 *           Michael J. Mahon - May 5, 1996
>115 *           Revised Oct 06, 2008
>116 *
>117 *           Copyright (c) 1996, 2004, 2008
>118 *
>119 * SERVER continually listens to the net, receiving all
>120 * packets, and responding to control packets directed
>121 * to 'self'.  If a key is pressed or a request handled,
>122 * SERVER returns.  C = 0 if count expired and is set as
>123 * the request server left it if a request was handled.
>124 *
>125 * To minimize missed polls, SERVER temporarily raises
>126 * RCVPKT's timeout to 20ms. from the normal value equal
>127 * to the minimum arbitration time.
>128 *
>129 * For every request code, there is a corresponding
>130 * server routine.  SERVER invokes these routines to
>131 * satisfy the service requests it receives.  Upon entry
>132 * to 'xxxSRV', C = 0 and (X) = (rbuf+rqmd).
>133 *
>134 * To ensure that the next packet received is the start
>135 * packet of a request protocol, it is necessary to wait
>136 * for the net to be idle or locked for at least the min
>137 * arbitration time before receiving a request.  (Note
>138 * that broadcast requests begin with the network in a
>139 * locked state.)
>140 *
>141 * The entry point 'svrxkbd' is provided for 'servelp',
>142 * which ignores keyboard input.
>143 *
>144 *****
>145

```

```

9403: AD 10 C0 >146 svrxkbd  lda  kbstroke  ; Ignore any keypress
>147
9406: A9 08 >148 SERVER  lda  #idleto  ; While polling, raise
9408: 8D 52 91 >149          sta  tolim    ; RCVPKT timeout to 20ms.
940B: A2 3A >150 :resync ldx  #servegap ; Delay min arb time
940D: CD E8 C0 >151          cmp  zipslow  ; Zip Chip to 1MHz mode.
9410: AC 62 C0 >152          ldy  drecv    ; Sample net state.
9413: 98 >153 :waitidl tya
9414: 4D 62 C0 >154          eor  drecv    ; Has net changed?
9417: 30 F2 >155          bmi  :resync ; -Yes, restart timing.
9419: CA >156          dex                    ; -No, count it down.
941A: D0 F7 >157          bne  :waitidl ; -Keep waiting.
941C: AD 00 C0 >158 :serve  lda  keybd    ; Check if key pressed.
941F: 30 75 >159          bmi  :exit    ; -Yes, return.
9421: 20 00 99 >160          jsr  RCVCTL   ; Receive ctl pkt to 'rbuf'
9424: B0 69 >161          bcs  :err      ; -Timeout or Cksum err.

```

```

9426: AD 46 91 >162      lda    rbuf+frmc    ; -Cksum OK, verify that
9429: 49 FF          >163      eor    #$FF        ; complement of 'frmc'
942B: CD 48 91 >164      cmp    rbuf+frm     ; is equal to 'frm'.
942E: D0 55          >165      bne    :frmcerr    ; -No, count collisions.
9430: AD 47 91 >166      lda    rbuf+dst     ; -Yes, good packet.
9433: F0 2D          >167      beq    :bcastck    ; Broadcast packet OK?
9435: CD 3C 91 >168      cmp    self        ; Directed to us?
9438: D0 3A          >169      bne    :skip       ; -No, just keep time.
943A: AD 45 91 >170      :bcast lda    rbuf+rqmd   ; -Yes, get 'rqmd'
943D: AA            >171      tax                    ; and save in X.
943E: 29 07          >172      and    #modmask    ; Is the modifier
9440: C9 01          >173      cmp    #rm_REQ     ; a Request?
9442: D0 2D          >174      bne    ]PROTERR    ; -No, protocol error.
9444: 8A            >175      txa                    ; -Yes, check request.
9445: 29 F8          >176      and    #reqmask    ;
9447: F0 28          >177      beq    ]PROTERR    ; Code must be > 0
9449: C9 70          >178      cmp    #maxreq     ; and < maxreq.
944B: B0 24          >179      bcs    ]PROTERR    ; Invalid request.
944D: 4A            >180      lsr                    ; Req code is * 8,
944E: 4A            >181      lsr                    ; so divide by 4. (C=0)
944F: A8            >182      tay                    ; Index of service routine
          >183      movl6 service-2,y;address ; Set up address
9450: B9 88 93 >183      lda    service-2,y ; Move 2 bytes
9453: 85 FC          >183      sta    address
9455: B9 89 93 >183      lda    1+service-2,y
9458: 85 FD          >183      sta    1+address
          >183      eom
945A: A9 01          >184      lda    #reqto      ; Reset timeout to min
945C: 8D 52 91 >185      sta    tolim       ; arbitration time.
945F: 6C FC 00 >186      jmp    (address)   ; Jump to service routine.
          >187
9462: AD 45 91 >188      :bcastck lda    rbuf+rqmd   ; Ck broadcast valid..
9465: C9 49          >189      cmp    #r_BPOKE+rm_REQ ; BPOKE request?
9467: F0 D1          >190      beq    :bcast      ; -Yes, process request.
9469: C9 41          >191      cmp    #r_BCAST+rm_REQ ; Broadcast BCAST req?
946B: F0 CD          >192      beq    :bcast      ; -Yes.
946D: C9 39          >193      cmp    #r_BOOT+rm_REQ ; Broadcast BOOT req?
946F: F0 03          >194      beq    :skip       ; -Yes, ignore.
9471: 20 8F 99 >195      ]PROTERR jsr    PROTERR    ; Record protocol error
9474: CE 53 91 >196      :skip  dec    reqctr ; Enough requests seen?
9477: D0 92          >197      bne    :resync     ; -No, re-sync SERVER.
9479: A9 03          >198      lda    #reqpidle   ; -Yes, about 20ms used.
947B: 8D 53 91 >199      sta    reqctr      ; Reset counter.
947E: CE 50 91 >200      dec    servecnt    ; Enough iterations?
9481: F0 13          >201      beq    :exit       ; -Yes, return.
9483: D0 86          >202      bne    :resync     ; -No, re-sync SERVER.
          >203
          >204      :frmcerr incl16 frmcerr ; Count sync'd collisions.
9485: EE 5A 91 >204      inc    frmcerr     ; Increment 16-bit word.
9488: D0 03          >204      bne    *+5         ; - No carry.
948A: EE 5B 91 >204      inc    frmcerr+1   ; Propagate carry.
          >204      eom

```

```
948D: D0 E5      >205          bne      :skip          ; (always)
          >206
948F: D0 E3      >207      :err      bne      :skip          ; -Cksum error.
9491: CE 50 91   >208          dec      servcnt        ; -Timeout. Enough?
9494: D0 86      >209          bne      :serve         ; -No, keep serving.
9496: A9 01      >210      :exit     lda      #reqto        ; -Yes, restore normal
9498: 8D 52 91   >211          sta      tolim          ; request timeout,
949B: 18         >212          clc                       ; clear Carry
949C: 60         >213          rts                       ; and return.
```

```
>217 *-----*
>218 *           Broadcast Boot & Bcast Protocol           *
>219 *-----*
>220 *           Master                               Slaves           *
>221 *  =====                               ===== *
>222 *  Bxxx  REQ (addr,leng)  ====> *
>223 *                (800 cyc. delay) *
>224 *                Data  ====> *
>225 *                : *
>226 *                Data  ====> *
>227 *-----*
```

```

>229 *****
>230 *
>231 *           B O O T R E Q   &   B C A S T R E Q
>232 *
>233 *           Michael J. Mahon - May 14, 2004
>234 *           Revised Apr 28, 2010
>235 *
>236 *           Copyright (c) 2004, 2008, 2010
>237 *
>238 * Broadcast request for all waiting machines to receive
>239 * data of 'sbuf+len' length.
>240 *
>241 * BOOTREQ is handled by all machines awaiting boot.
>242 * The boot image following is loaded at 'sbuf+adr' and
>243 * control is passed to the boot image.
>244 *
>245 * BCASTREQ is handled by all machines awaiting BCAST
>246 * data. 'sbuf+adr' is the "tag" for the data following,
>247 * that is ignored or received by waiting machines based
>248 * on their state and the tag value.
>249 *
>250 * Since these requests are broadcast, they do not get
>251 * ACKs from their destination(s), but simply send their
>252 * data blindly. If errors occur, waiting machines will
>253 * continue to wait for good data, so verification of
>254 * proper operation must be handled separately.
>255 *
>256 * Because broadcast data is sent "open loop", and since
>257 * BCAST clients may require time to determine whether
>258 * and how they should receive the following data, these
>259 * protocols delay for 800 cycles between the request
>260 * and the sending of data.
>261 *
>262 * BOOTREQ & BCASTREQ do the following steps:
>263 *     1. Sets up the request
>264 *     2. Does a broadcast arbitration to seize the net
>265 *        and delay 20ms to resolve any collisions and
>266 *        allow slow pollers to reach their RCVPKT holds
>267 *     3. Sends the request, with address/tag and length
>268 *     4. Waits 800 cyc. for clients to prepare to
>269 *        receive the data (or not).
>270 *     5. Sends the boot code/data stream
>271 *
>272 *****
>273
949D: A9 39 >274 BOOTREQ lda #r_BOOT+rm_REQ
949F: D0 02 >275 bne ldoit ; (always)
>276
94A1: A9 41 >277 BCASTREQ lda #r_BCAST+rm_REQ ; BCAST request
94A3: 8D 3D 91 >278 ldoit sta sbuf+rqmd
94A6: 20 1A 97 >279 jsr BCASTARB ; Bcast arbitrate & lock bus
94A9: 20 26 98 >280 jsr SENDCTL ; Send the BOOT request.

```

```
94AC: 20 E9 98 >281      jsr   lasl=>a1    ; Local start address & length
94AF: A2 A0      >282      ldx   #800/5     ; Delay 800 cycles,
94B1: 4C AD 99 >283      jmp   DSENDLNG   ; send data and return.
```

```
>286
>287
>288
>289
>290
>291
>292
>293
```

```
>294 *-----*
>295 *                GETID Protocol                *
>296 *-----*
>297 *           Requester                           Server           *
>298 * ===== *
>299 * GETID  REQ (pseudo ID)  =====> *
>300 * * <===== GETID  ACK (ID) *
>301 * GETID  DACK (ID)      =====> *
>302 *-----*
```

```

>304 *****
>305 *
>306 *           G E T I D S R V
>307 *
>308 *           Michael J. Mahon - May 14, 2004
>309 *           Revised Aug 17, 2008
>310 *
>311 *           Copyright (c) 2004, 2008
>312 *
>313 * Service machine 'rbuf+frm's request to allocate a new
>314 * machine ID (if 'rbuf+frm' is a pseudo-ID).
>315 *
>316 * The new ID is sent in the ACK packet.  GETIDSRV
>317 * requires a DACK packet from the newly allocated
>318 * machine ID before the new allocation is committed.
>319 *
>320 * GETIDSRV is unique in that it returns control
>321 * directly to SERVER (rather than SERVER's caller), so
>322 * that all GETID requests are processed before SERVER
>323 * returns.
>324 *
>325 * GETIDSRV does the following steps:
>326 *     1. If a pseudo-ID was received, it finds the next
>327 *        available machine ID.
>328 *     2. Sends the new (or current) ID in the ACK packet
>329 *     3. Receives the DACK and marks the ID allocated.
>330 *     4. Gives control back to SERVER.
>331 *
>332 *****
>333

```

```

94B4: AE 48 91 >334 GETIDSRV ldx  rbuf+frm  ; Look at requester's ID
94B7: 10 0E   >335         bpl  :ack      ; -it's real, just ACK.
94B9: A2 02   >336         ldx  #2       ; -pseudo, find new one.
94BB: BD 5D 91 >337 :search lda  idtable,x ; Find lowest
94BE: F0 07   >338         beq  :ack      ; unused ID.
94C0: E8     >339         inx
94C1: E0 20   >340         cpx  #maxid+1
94C3: 90 F6   >341         bcc  :search
94C5: B0 19   >342         bcs  :exit      ; Table overflow!
>343
94C7: 8E 41 91 >344 :ack   stx  sbuf+adr  ; Send ID to requester
94CA: 20 12 98 >345         jsr  SENDACK
94CD: AD 41 91 >346         lda  sbuf+adr  ; Expect new ID
94D0: 8D 3F 91 >347         sta  sbuf+dst  ; in DACK.
94D3: 20 9E 96 >348         jsr  RCVDACK
94D6: B0 08   >349         bcs  :exit      ; -Error, don't allocate.
94D8: AE 3F 91 >350         ldx  sbuf+dst  ; -OK.
94DB: A9 01   >351         lda  #1
94DD: 9D 5D 91 >352         sta  idtable,x ; Allocate the ID
94E0: 4C 06 94 >353 :exit  jmp  SERVER   ; Go back to SERVER.

```

```
58          put    PEEKPOKECALL
>2      *-----*
>3      *          Requester                      Server          *
>4      *  =====                               =====       *
>5      *  PEEK   REQ (addr,leng)  =====>                    *
>6      *                                          <===== PEEK   ACK          *
>7      *                                          <===== Data (if >4 bytes) *
>8      *                                          :                          *
>9      *                                          <===== Data          *
>10     *-----*
```



```

>14 *****
>15 *
>16 * P E E K R E Q *
>17 *
>18 * Michael J. Mahon - May 5, 1996 *
>19 * Revised May 21, 2008 *
>20 *
>21 * Copyright (c) 1996, 2008 *
>22 *
>23 * Request machine 'sbuf+dst' to send 'sbuf+len' bytes *
>24 * at its 'sbuf+adr', and put them at location 'locaddr'. *
>25 *
>26 * PEEKREQ, like other requests, will retry the request *
>27 * in case of error, up to 'maxreqrt' times. If errors *
>28 * persist, it will return with Carry set. *
>29 *
>30 * PEEKREQ does the following steps: *
>31 * 1. Make the PEEK request (and receive the ACK) *
>32 * 2. Receive 'sbuf+len' bytes of data into 'locaddr' *
>33 * 3. Retry in case of error up to 'maxreqrt' times *
>34 *
>35 *****
>36

```

```

94E3: A9 03 >37 PEEKREQ lda #maxreqrt ; Set request retry
94E5: 8D 54 91 >38 sta reqretry ; counter.
94E8: A9 08 >39 :retry lda #r_PEEK ; Send PEEK request.
94EA: 20 42 96 >40 jsr REQUEST
94ED: B0 1E >41 bcs :failed
94EF: 20 E9 98 >42 jsr lasl=>al ; Set up address/length
94F2: A5 FF >43 lda length+1 ; If length
94F4: D0 12 >44 bne :long ; is >255 bytes, or
94F6: A4 FE >45 ldy length ; if length is
94F8: F0 19 >46 beq :done ; (length = 0!)
94FA: C0 05 >47 cpy #5 ; > 4 bytes,
94FC: B0 0A >48 bcs :long ; receive multiple pkts.
94FE: 88 >49 dey ; Move short response
94FF: B9 49 91 >50 :short lda rbuf+adr,y ; to local data address.
9502: 91 FC >51 sta (address),y
9504: 88 >52 dey
9505: 10 F8 >53 bpl :short
9507: 60 >54 rts ; ...and return.
>55
9508: 20 CF 99 >56 :long jsr RCVLONG ; Receive multiple packets
950B: 90 06 >57 bcc :done ; No problem.
950D: CE 54 91 >58 :failed dec reqretry ; Dec request retry count
9510: D0 D6 >59 bne :retry ; Try until OK or exhausted,
9512: 38 >60 sec ; then return with C set.
9513: 60 >61 :done rts

```

```

>65 *****
>66 *
>67 *           P E E K S R V
>68 *
>69 *           Michael J. Mahon - May 5, 1996
>70 *           Revised May 21, 2008
>71 *
>72 *           Copyright (c) 1996, 2005, 2008
>73 *
>74 * Service machine 'rbuf+frm's request to send 'rbuf+len'*
>75 * bytes of data from our 'rbuf+adr'.
>76 *
>77 * PEEKSRV does the following steps:
>78 *     1. Check 'rbuf+len' for a 1..4 byte request
>79 *     2. Send the ACK packet (with data, if short)
>80 *     3. If long, send multiple response packets
>81 *
>82 *****
>83

```

```

9514: 20 98 99 >84 PEEKSRV jsr rarl=>al ; Set address/length.
9517: A5 FF >85 lda length+1 ; Check for long response
9519: D0 14 >86 bne :long
951B: A4 FE >87 ldy length ; Check for < 5 bytes.
951D: F0 0D >88 beq :nullreq ; length = 0.
951F: C0 05 >89 cpy #5
9521: B0 0C >90 bcs :long ; - No, longer.
9523: 88 >91 dey ; - Yes, move response
9524: B1 FC >92 :short lda (address),y ; data into ACK packet.
9526: 99 41 91 >93 sta sbuf+adr,y
9529: 88 >94 dey
952A: 10 F8 >95 bpl :short
952C: 4C 12 98 >96 :nullreq jmp SENDACK ; Send ACK with response.
>97
952F: 20 12 98 >98 :long jsr SENDACK ; ACK the request.
9532: A2 1C >99 ldx #140/5 ; Allow requester to receive.
9534: 4C AD 99 >100 jmp DSENDLNG ; Send long response.

```

```

>102 *-----*
>103 *           Requester                               Server           *
>104 * =====
>105 * PEEKINC REQ (addr,inc) <====>
>106 *                               <==== PEEKINC ACK (oldval)
>107 *-----*
>108
>111 *****
>112 *
>113 *           P K I N C R E Q ,   P K P O K R E Q
>114 *
>115 *           Michael J. Mahon - Nov 05, 2004
>116 *
>117 *           Copyright (c) 2004, 2010
>118 *
>119 * Request machine 'sbuf+dst' to return 2 bytes at its
>120 * 'sbuf+adr', then increment that value by 'sbuf+len'
>121 * (PEEKINC), or set the value to 'sbuf+len' (PEEKPOKE).
>122 * The returned, unchanged value is at 'rbuf+len'.
>123 *
>124 * These requests, like others, will retry the request
>125 * in case of error, up to 'maxreqrt' times. If errors
>126 * persist, SIMPLREQ will return with Carry set.
>127 *
>128 * PEEKINC and PEEKPOKE do the following steps:
>129 *     1. Make the request (and receive the ACK)
>130 *     2. Retry in case of error up to 'maxreqrt' times
>131 *
>132 *****
>133
9537: A9 50 >134 PKINCREQ lda    #r_PKINC    ; Send PEEKINC request
9539: D0 02 >135         bne    SIMPLREQ    ; (always)
>136
953B: A9 58 >137 PKPOKREQ lda    #r_PKPOK    ; Send PEEKPOKE request
>138
953D: 8D 3D 91 >139 SIMPLREQ sta    sbuf+rqmd    ; Save request type.
9540: A9 03 >140         lda    #maxreqrt    ; Set request retry
9542: 8D 54 91 >141         sta    reqretry    ; counter.
9545: AD 3D 91 >142 :retry  lda    sbuf+rqmd    ; Send request.
9548: 20 42 96 >143         jsr    REQUEST
954B: 90 06 >144         bcc    :done      ; Done if no error.
954D: CE 54 91 >145         dec    reqretry    ; Dec request retry count
9550: D0 F3 >146         bne    :retry    ; Try until OK or exhausted,
9552: 38 >147         sec          ; then return with C set.
9553: 60 >148         :done    rts

```

```

>152 *****
>153 *
>154 *           P K I N C S R V ,   P K P O K S R V
>155 *
>156 *           Michael J. Mahon - Nov 05, 2004
>157 *           Revised Apr 30, 2010
>158 *
>159 *           Copyright (c) 2004, 2008
>160 *
>161 * Service machine 'rbuf+frm's request to send 2 bytes
>162 * at our 'rbuf+adr', then increment value by 'rbuf+len'
>163 * (PEEKINC) or store 'rbuf+len' as new value (PEEKPOKE).
>164 *
>165 * The PEEKINC and PEEKPOKE requests are "network atomic"
>166 * read-modify-write primitives for synchronization and
>167 * allocation operations.
>168 *
>169 * PKINCSRV and PKPOKSRV do the following steps:
>170 *     1. Save initial 2-byte value in ACK buffer
>171 *     2. If PKINCSRV: Increment the value by 'rbuf+len'
>172 *     3. If PKPOKSRV: Set the value to 'rbuf+len'.
>173 *     4. Send the ACK packet with the initial value.
>174 *
>175 *****
>176

```

```

9554: A9 FF >177 PKINCSRV lda    #$FF      ; Set mask to "increment"
9556: D0 02 >178         bne    pkxxxxsrv ; and go do it.
>179
9558: A9 00 >180 PKPOKSRV lda    #0        ; Set mask to "move"
955A: 85 EC >181 pkxxxxsrv sta    ckbyte
955C: 20 A2 99 >182         jsr    ra=>a      ; Set up data address
955F: A0 00 >183         ldy    #0
9561: 18 >184         clc
9562: B1 FC >185 :movinc lda    (address),y ; Get original value
9564: 99 43 91 >186         sta    sbuf+len,y ; and save it for ACK.
9567: 25 EC >187         and    ckbyte      ; ($FF=inc, $00=move)
9569: 79 4B 91 >188         adc    rbuf+len,y
956C: 91 FC >189         sta    (address),y ; Replace with new value.
956E: C8 >190         iny
956F: 98 >191         tya          ; Don't disturb carry.
9570: 49 02 >192         eor    #2        ; Done?
9572: D0 EE >193         bne    :movinc    ; -No, go again.
9574: 4C 12 98 >194         jmp    SENDACK     ; -Yes, send ACK with value.

```



```

>211 *****
>212 *
>213 *      P O K E R E Q ,   R U N R E Q ,   B R U N R E Q
>214 *
>215 *              Michael J. Mahon - May 11, 1996
>216 *              Revised Sep 25, 2008
>217 *
>218 *              Copyright (c) 1996, 2004, 2008
>219 *
>220 * Request machine 'sbuf+dst' to store 'sbuf+len' bytes
>221 * at its 'sbuf+adr', and send them from our location
>222 * 'locaddr'.
>223 *
>224 * These requests, like others, will retry the request
>225 * in case of error, up to 'maxreqrt' times.  If errors
>226 * persist, it will return with Carry set.
>227 *
>228 * POKEREQ, RUNREQ, and BRUNREQ do the following steps:
>229 *     1. Make the request (and receive the ACK)
>230 *     2. Send 'sbuf+len' bytes of data from 'locaddr'
>231 *     3. Receive DATA ACK response
>232 *     4. Retry in case of error up to 'maxreqrt' times
>233 *
>234 *****
>235

```

```

9577: A9 60 >236 RUNREQ  lda    #r_RUN    ; Send RUN request.
9579: D0 06 >237         bne    setreq   ; (always)
>238
957B: A9 68 >239 BRUNREQ lda    #r_BRUN   ; Send BRUN request.
957D: D0 02 >240         bne    setreq   ; (always)
>241
957F: A9 10 >242 POKEREQ lda    #r_POKE   ; Send POKE request.
9581: 8D 3D 91 >243 setreq  sta    sbuf+rqmd ; Set request code
9584: A2 03 >244         ldx    #maxreqrt ; Set request retry
9586: 8E 54 91 >245         stx    reqretry ; counter.
9589: AD 3D 91 >246 :retry  lda    sbuf+rqmd ; Recover request code
958C: 20 42 96 >247         jsr    REQUEST
958F: B0 0B >248         bcs    :failed
9591: 20 E9 98 >249         jsr    lasl=>al  ; Set up address/length.
9594: 20 B0 99 >250         jsr    SENDLONG ; Send multiple packets.
9597: 20 9E 96 >251         jsr    RCVDACK  ; Receive DATA ACK packet.
959A: 90 06 >252         bcc    :done    ; -OK, return.
959C: CE 54 91 >253 :failed dec    reqretry ; Dec request retry count
959F: D0 E8 >254         bne    :retry   ; Try until OK or exhausted,
95A1: 38 >255         sec     ; then return with C set.
95A2: 60 >256 :done  rts

```

```

>260 *****
>261 *
>262 *   P O K E S R V ,   R U N S R V ,   B R U N S R V   *
>263 *
>264 *           Michael J. Mahon - May 11, 1996   *
>265 *                   Revised Jan 24, 2008   *
>266 *
>267 *           Copyright (c) 1996, 2008, 2009   *
>268 *
>269 *   Service machine 'rbuf+frm's request to poke 'rbuf+len'*
>270 *   bytes of data to our 'rbuf+adr'.   *
>271 *
>272 *   If RUNSRV, initialize Applesoft and RUN the BASIC   *
>273 *   program transferred.  (Address must be > $800.)   *
>274 *
>275 *   If BRUNSRV, CALL the code transferred with (A,X) set   *
>276 *   to the code's load address.   *
>277 *
>278 *   POKESRV, RUNSRV, and BRUNSRV do the following steps:   *
>279 *       1. If RUNSRV: lock net, save CSW/KSW hooks, and   *
>280 *           coldstart Applesoft.   *
>281 *       2. Send the ACK packet   *
>282 *       3. Receive multiple packets to 'rbuf+adr'   *
>283 *       4. If data received OK, send DATA ACK packet   *
>284 *       5. If BRUNSRV: Set (A,X) to address & JMP to code.   *
>285 *       6. If RUNSRV: Init Applesoft ptrs, fix up links,   *
>286 *           restore CSW/KSW hooks, and RUN the program.   *
>287 *
>288 *****
>289
>291 savhooks equ    $2FC           ; Save hooks at end of page 2
95A3: 16 96 >292 sethooks dw    rts           ; For COLDSTRT and FIXLINKS
95A5: BC 95 >293          dw    POKESRV        ; use hooks to retain control.
>294
95A7: 8D 5B C0 >295 RUNSRV   sta    dsend+1        ; Lock net for coldstart
95AA: A2 03 >296          ldx    #3           ; Save and set CSW/KSW hooks
95AC: B5 36 >297 :saveset lda    CSW,x         ; to <rts,POKESRV> to retain
95AE: 9D FC 02 >298          sta    savhooks,x    ; control after coldstart.
95B1: BD A3 95 >299          lda    sethooks,x
95B4: 95 36 >300          sta    CSW,x
95B6: CA >301          dex
95B7: 10 F3 >302          bpl    :saveset
95B9: 4C 00 E0 >303          jmp    COLDSTRT        ; BASIC coldstart.
>305 BRUNSRV equ    *
95BC: 20 12 98 >306 POKESRV  jsr    SENDACK        ; ACK the request.
95BF: 20 98 99 >307          jsr    rarl=>al        ; Set up address/length.
95C2: 20 CF 99 >308          jsr    RCVLONG        ; Receive long data message.
95C5: B0 4F >309          bcs    rts           ; Receive error.
>310          delay 40          ; Allow requester to receive.
95C7: A2 08 >310          ldx    #40/5          ; (5 cycles per iteration)
95C9: CA >310          ]delay dex
95CA: D0 FD >310          bne    ]delay

```

```

>310          eom
95CC: A9 03   >311   lda   #rm_DACK   ; Send DATA ACK
95CE: 20 14 98 >312   jsr   SENDRSP   ; packet.
95D1: AD 45 91 >313   lda   rbuf+rqmd ; Recover request
95D4: C9 11   >314   cmp   #r_POKE+rm_REQ ; POKE?
95D6: F0 3D   >315   beq   :ok       ; -Yes, return.
95D8: C9 69   >316   cmp   #r_BRUN+rm_REQ ; -No, BRUN?
95DA: F0 5D   >317   beq   docall    ; -Yes, do CALL.
95DC: AD CF 03 >321   lda   nadapage  ; -No, RUN. Set HIMEM to
95DF: 85 74   >322   sta   HIMEM+1  ; NadaNet load page.
95E1: 85 70   >323   sta   FRETOP+1
95E3: 18     >324   clc
95E4: AD 49 91 >325   lda   rbuf+adr  ; Set PSTART to start
95E7: 85 67   >326   sta   PSTART   ; addr and VARTAB to
95E9: 6D 4B 91 >327   adc   rbuf+len  ; end of program.
95EC: 85 69   >328   sta   VARTAB
95EE: AD 4A 91 >329   lda   rbuf+adr+1
95F1: 85 68   >330   sta   PSTART+1
95F3: 6D 4C 91 >331   adc   rbuf+len+1
95F6: 85 6A   >332   sta   VARTAB+1
          >333   movl6 #:run;KSW ; Retain control after
95F8: A9 03   >333   lda   #:run     ; Move 2 bytes
95FA: 85 38   >333   sta   KSW
95FC: A9 96   >333   lda   #:run/$100 ; high byte of immediate
95FE: 85 39   >333   sta   1+KSW
          >333   eom
9600: 4C F2 D4 >334   jmp   FIXLINKS ; fixing up prog links.
9603: A2 04   >335   :run  ldx   #4 ; Restore CSW/KSW hooks.
9605: BD FB 02 >336   :restore lda savhooks-1,x
9608: 95 35   >337   sta   CSW-1,x
960A: CA     >338   dex
960B: D0 F8   >339   bne   :restore
960D: 8A     >340   txa   ; Set byte preceding
960E: 81 B8   >341   sta   (TXTPTR,x) ; program to zero,
9610: 85 D8   >342   sta   ONERR ; clear ONERR flag, and
9612: 4C 66 D5 >343   jmp   RUNPROG ; RUN the Applesoft prog.
          >345
9615: 18     >346   :ok  clc   ; Good return.
9616: 60     >347   rts   ; Return.

```



```

>349 *-----*
>350 *           Requester                               Server           *
>351 * =====
>352 * BPOKE REQ (addr,val)  ==>
>353 *                                     (Broadcast, No ACK)
>354 *-----*
>355
>358 *****
>359 *
>360 *           B P O K E R E Q
>361 *
>362 *           Michael J. Mahon - Nov. 04, 2004
>363 *           Revised Aug 20, 2008
>364 *
>365 *           Copyright (c) 2004, 2008
>366 *
>367 * Broadcast request to all serving machines to store 2
>368 * bytes in 'sbuf+len' at address 'sbuf+adr'.
>369 *
>370 * BPOKE, unlike most requests, is broadcast, and so
>371 * is not acknowledged by any receiver. To eliminate
>372 * the chance of collision, it holds the bus locked for
>373 * 20ms after arbitration, then sends the request packet.*
>374 * This allows enough time for any colliding sender to
>375 * send its request and re-arbitrate while the bus is
>376 * locked, so that there is no contention when the BPOKE
>377 * request is finally sent.
>378 *
>379 * BPOKEREQ does the following steps:
>380 *     1. Broadcast arbitrate and lock the bus
>381 *     2. Set up BPOKE request
>382 *     3. Send the BPOKE request packet.
>383 *
>384 *****
>385
9617: 20 1A 97 >386 BPOKEREQ jsr   BCASTARB   ; Bcast arbitrate & lock bus
961A: A9 49   >387         lda   #r_BPOKE+rm_REQ ; Set up BPOKE request.
961C: 8D 3D 91 >388         sta  sbuf+rqmd
961F: 4C 26 98 >389         jmp   SENDCTL     ; Send the request.

```

```

>393 *****
>394 *
>395 *           B P O K E S R V
>396 *
>397 *           Michael J. Mahon - Nov 05, 2004
>398 *           Revised May 21, 2008
>399 *
>400 *           Copyright (c) 2004, 2008
>401 *
>402 * Service machine 'rbuf+frm's request to poke 2 bytes
>403 * of data in 'rbuf+len' to our 'rbuf+adr'.
>404 *
>405 * BPOKESRV does the following:
>406 *     1. Move 'rbuf+len' to memory at 'rbuf+adr'.
>407 *
>408 *****
>409

```

```

9622: 20 A2 99 >410 BPOKESRV jsr    ra=>a        ; Set up pointer
9625: A0 01    >411          ldy    #1          ; and move 2 bytes.
9627: B9 4B 91 >412 :move   lda    rbuf+len,y
962A: 91 FC    >413          sta    (address),y
962C: 88      >414          dey
962D: 10 F8    >415          bpl    :move
962F: 18      >416          clc
9630: 60      >417          rts                ; All done.

```

```

>419 *-----*
>420 *           Requester                               Server           *
>421 * =====
>422 * CALL  REQ (addr,A,X)  <====>
>423 *                               <==== CALL  ACK
>424 *-----*
>425
>428 *****
>429 *
>430 *           C A L L R E Q
>431 *
>432 *           Michael J. Mahon - May 11, 1996
>433 *           Revised Apr 30, 2010
>434 *
>435 *           Copyright (c) 1996, 2004, 2010
>436 *
>437 * Request machine 'sbuf+dst' to call a subroutine at
>438 * address 'sbuf+adr' with parameters A = 'sbuf+len' and
>439 * X = 'sbuf+len+1'.
>440 *
>441 * CALLREQ jumps to SIMPLREQ to retry the request in case*
>442 * of error, up to 'maxreqrt' times.  If errors persist, *
>443 * SIMPLREQ will return to the caller with Carry set.
>444 *
>445 * CALLREQ does the following steps:
>446 *     1. Make the CALL request (and receive the ACK)
>447 *     2. Retry in case of error up to 'maxreqrt' times
>448 *
>449 *****
>450
9631: A9 18 >451 CALLREQ  lda  #r_CALL    ; Send CALL request and
9633: 4C 3D 95 >452          jmp  SIMPLREQ    ; receive ACK (or error).

```

```

>456 *****
>457 *
>458 *           C A L L S R V
>459 *
>460 *           Michael J. Mahon - May 11, 1996
>461 *           Revised Sep 25, 2008
>462 *
>463 *           Copyright (c) 1996, 2004, 2008
>464 *
>465 * Service machine 'rbuf+frm's request to call a
>466 * subroutine at our 'rbuf+adr' with parameters
>467 * A = 'rbuf+len' and X = 'rbuf+len+1'.  Flags are set
>468 * according to the value of A.
>469 *
>470 * Note that when the subroutine returns, it returns to
>471 * whoever called SERVER.
>472 *
>473 * CALLSRV does the following steps:
>474 *     1. Send the ACK packet
>475 *     2. Load parameters from 'rbuf+len' into A and X
>476 *     3. Call subroutine at 'rbuf+adr'
>477 *
>478 *****
>479
9636: 20 12 98 >480 CALLSRV jsr SENDACK ; ACK the request.
9639: AE 4C 91 >481 docall ldx rbuf+len+1 ; Set X parameter
963C: AD 4B 91 >482 lda rbuf+len ; and A parameter, and
963F: 6C 49 91 >483 jmp (rbuf+adr) ; Jump to requested address.

```

```

>487 *****
>488 *
>489 *           R E Q U E S T
>490 *
>491 *           Michael J. Mahon - April 20, 2004
>492 *           Revised Aug 17, 2008
>493 *
>494 *           Copyright (c) 1996, 2004, 2008
>495 *
>496 * Handle request protocol for the request in A & 'sbuf'.
>497 *
>498 * Retry the protocol for up to 'reqtime' ms. (up to
>499 * 'retrylim' times). If successful, return with valid
>500 * response in 'rbuf' and Carry clear.
>501 *
>502 * If request timed out, return with Carry set and A=0.
>503 *
>504 * If NAK received, return with Carry set and A>0.
>505 *
>506 * REQUEST performs the following steps:
>507 *     1. Complete control pkt in 'sbuf' (request in A)
>508 *     2. Arbitrate for the use of the bus
>509 *     3. Send the request specified in 'sbuf'
>510 *     4. Receive the control response into 'rbuf'
>511 *     5. Check 'rbuf' for a valid, expected response
>512 *     6. Retry steps 2 to 5 up to 'retrylim' times
>513 *     7. When ACKed, NAKed, or timed-out, return
>514 *
>515 *****
>516

```

```

9642: 09 01 >517 REQUEST ora #rm_REQ ; Add REQ modifier and
9644: 8D 3D 91 >518 sta sbuf+rqmd ; Store request code.
9647: AD 4F 91 >519 lda retrylim ; Init retry counter.
964A: 8D 55 91 >520 sta retrycnt
964D: AD 55 91 >521 :retry lda retrycnt ; Timed out?
9650: F0 4A >522 beq :err ; -Yes, return w/ C set, A=0
9652: CE 55 91 >523 dec retrycnt ; Dec retry counter.
9655: 20 00 98 >524 jsr ARBTRATE ; Arbitrate for & lock bus
9658: 20 26 98 >525 jsr SENDCTL ; Send request in 'sbuf'.
965B: 20 00 99 >526 jsr RCVCTL ; Receive response in 'rbuf'.
965E: 90 0D >527 bcc :ok ; -Clean packet received.
>528 dlyms reqdelay ; delay a few ms.
9660: A0 11 >528 ldy #reqdelay ; Delay lms. per iteration
>528 ]dly delay 1020-4 ; Cycles per ms. - 4
9662: A2 CB >528 ldx #1020-4/5 ; (5 cycles per iteration)
9664: CA >528 ]delay dex
9665: D0 FD >528 bne ]delay
>528 eom
9667: 88 >528 dey
9668: D0 F8 >528 bne ]dly
>528 eom
966A: 4C 4D 96 >529 jmp :retry ; and try again...

```

```

>530
966D: AD 47 91 >531 :ok      lda    rbuf+dst    ; Message received, is
9670: CD 3C 91 >532          cmp    self        ; it for us?
9673: D0 1D          >533          bne    :protterr  ; -No, error.
9675: AD 3F 91 >534          lda    sbuf+dst    ; -Yes. Is it from
9678: CD 48 91 >535          cmp    rbuf+frm    ; our destination?
967B: D0 15          >536          bne    :protterr  ; -No. Protocol error.
967D: AD 3D 91 >537          lda    sbuf+rqmd   ; -Yes. Is the
9680: 29 F8          >538          and    #reqmask    ; modifier field
9682: 09 02          >539          ora    #rm_ACK     ; 'ACK'?
9684: CD 45 91 >540          cmp    rbuf+rqmd   ; as expected?
9687: F0 0F          >541          beq    :good       ; -Yes, good response!
9689: 29 F8          >542          and    #reqmask    ; -No, construct
968B: 09 04          >543          ora    #rm_NAK     ; the 'NAK' value.
968D: CD 45 91 >544          cmp    rbuf+rqmd   ; Is it a NAK?
9690: F0 08          >545          beq    :nakexit    ; -Yes, return w/ C set, A=1
9692: 20 8F 99 >546 :protterr jsr    PROTERR    ; -No, count protocol errors.
9695: 4C 4D 96 >547          jmp    :retry      ; and try again...
>548
9698: 18          >549 :good   clc          ; Signal good ACK
9699: 60          >550          rts          ; and return.
>551
969A: A9 01          >552 :nakexit lda    #1        ; Signal NAK
969C: 38          >553 :err    sec          ; Signal error
969D: 60          >554          rts          ; and return.

```

```

>558 *****
>559 *
>560 *           R C V D A C K
>561 *
>562 *           Michael J. Mahon - Apr 19, 2004
>563 *           Revised Aug 17, 2008
>564 *
>565 *           Copyright (c) 2004, 2008
>566 *
>567 * Receive DATA ACK packet. Require a good cksum,
>568 * addressed to us, response req = sent req. If all OK,
>569 * return with Carry clear, else with Carry set.
>570 *
>571 *****
>572

```

```

969E: 20 00 99 >573 RCVDACK jsr RCVCTL ; Receive response packet.
96A1: B0 1E >574 bcs :err ; Cksum error or timeout.
96A3: AD 47 91 >575 :ok lda rbuf+dst ; Is packet for us?
96A6: CD 3C 91 >576 cmp self
96A9: D0 18 >577 bne :proterr ; -No, protocol error.
96AB: AD 48 91 >578 lda rbuf+frm ; Was it sent by receiver?
96AE: CD 3F 91 >579 cmp sbuf+dst
96B1: D0 10 >580 bne :proterr ; -No, protocol error.
96B3: AD 3D 91 >581 lda sbuf+rqmd ; Construct sent req
96B6: 29 F8 >582 and #reqmask ; with the expected
96B8: 09 03 >583 ora #rm_DACK ; 'DACK' modifier.
96BA: CD 45 91 >584 cmp rbuf+rqmd ; Does it match?
96BD: D0 04 >585 bne :proterr ; -No, protocol error.
96BF: 18 >586 clc ; -Yes, clear Carry
96C0: 60 >587 :return rts ; and return.
>588
96C1: D0 FD >589 :err bne :return ; Cksum error return.
96C3: 38 >590 :proterr sec ; Return with C set,
96C4: 4C 8F 99 >591 jmp PROTERR ; after counting error.

```

```

59          put    PUTMGETM
>1 *****
>2 *
>3 *          Message Server
>4 *
>5 *          Michael J. Mahon - April 20, 2004
>6 *          Revised May 21, 2008
>7 *
>8 *          Copyright (c) 2004, 2005, 2008
>9 *
>10 *         Client Request Routines
>11 *         Put Message Request
>12 *         Get Message Request
>13 *
>14 *         Server Definitions
>15 *         Message Page Table
>16 *         Message Class Table
>17 *         Message Buffers (pages)
>18 *
>19 *         Server Routines (w/ Monitor)
>20 *         Put Message Server
>21 *         Get Message Server
>22 *
>23 *         Utility Routines
>24 *         Look Up class in Message Table
>25 *
>26 *****
>27
>28 *-----*
>29 *          Requester          Server
>30 *          =====
>31 *          PUTMSG REQ (class,leng) ==>
>32 *          (lock)          :
>33 *          (<==== PUTMSG NAK if no space)
>34 *
>35 *          <==== PUTMSG ACK
>36 *          Data < 256 bytes ==>
>37 *          <==== PUTMSG DACK
>38 *-----*
>39 *          GETMSG REQ (class)  ==>
>40 *          (lock)          :
>41 *          (<==== GETMSG NAK if no msg)
>42 *
>43 *          <==== GETMSG ACK (class,leng)
>44 *          <==== Data < 256 bytes
>45 *          GETMSG DACK        ==>
>46 *-----*

```



```

>49 *****
>50 *
>51 *           P U T M R E Q
>52 *
>53 *           Michael J. Mahon - April 17, 2004
>54 *           Revised May 21, 2008
>55 *
>56 *           Copyright (c) 2004, 2008
>57 *
>58 * Request message server (at 'sbuf+dest') to accept a
>59 * message of class 'sbuf+adr' and length 'sbuf+len'
>60 * at our local address 'locaddr'.
>61 *
>62 * PUTMREQ will retry the request in case of timeout or
>63 * checksum errors up to 'maxreqrt' times.  If errors
>64 * persist, it returns with C set and A=0.
>65 *
>66 * If the server NAKs the request for lack of space,
>67 * PUTMREQ returns with C set and A=1.
>68 *
>69 * PUTMREQ does the following steps:
>70 *     1. Make the PUTMSG request
>71 *     2. If server NAKs, return with C set and A=1.
>72 *     3. Send 'sbuf+len'-byte message from 'locaddr'
>73 *     4. Receive DATA ACK packet
>74 *     5. Retry in case of error up to 'maxreqrt' times
>75 *     6. If unsuccessful, return with C set and A=0.
>76 *
>77 *****
>78

```

```

96C7: A9 03 >79 PUTMREQ lda #maxreqrt ; Set request retry
96C9: 8D 54 91 >80 sta reqretry ; counter.
96CC: A9 20 >81 :retry lda #r_PUTMSG ; Send PUTMSG request.
96CE: 20 42 96 >82 jsr REQUEST
96D1: B0 0D >83 bcs :failed
96D3: 20 E9 98 >84 jsr lasl=>a1 ; Set up address/length
96D6: 20 B0 99 >85 jsr SENDLONG ; and send message.
96D9: 20 9E 96 >86 jsr RCVDACK ; Receive DATA ACK packet.
96DC: 90 0A >87 bcc :done ; -All OK.
96DE: A9 00 >88 lda #0 ; Not a NAK error
96E0: D0 05 >89 :failed bne :nakexit
96E2: CE 54 91 >90 :cksumer dec reqretry ; Dec request retry count
96E5: D0 E5 >91 bne :retry ; Try until OK or exhausted,
96E7: 38 >92 :nakexit sec ; then return with C set.
96E8: 60 >93 :done rts

```

```

>95 *****
>96 *
>97 *           G E T M R E Q
>98 *
>99 *           Michael J. Mahon - April 19, 2004
>100 *                Revised May 21, 2008
>101 *
>102 *                Copyright (c) 2004, 2008
>103 *
>104 * Request message server (at 'sbuf+dst') to deliver
>105 * the first message of class 'sbuf+adr' to our address
>106 * 'locaddr', actual length in 'rbuf+len' after ACK.
>107 *
>108 * GETMREQ will retry the request in case of timeout or
>109 * checksum errors up to 'maxreqrt' times. If errors
>110 * persist, it returns with C set and A=0.
>111 *
>112 * If the server NAKs the request because the message
>113 * queue is empty, GETMREQ returns with C set and A=1.
>114 *
>115 * GETMREQ does the following steps:
>116 *     1. Make the GETMSG request
>117 *     2. If server NAKs, return with C set and A=1.
>118 *     3. Receive 'rbuf+len'-byte message to 'locaddr'
>119 *     4. If no error, send DATA ACK packet
>120 *     5. Retry in case of error up to 'maxreqrt' times
>121 *     6. If unsuccessful, return with C set and A=0.
>122 *
>123 *****
>124

```

```

96E9: A9 03   >125 GETMREQ  lda    #maxreqrt  ; Set request retry
96EB: 8D 54 91 >126          sta    reqretry  ; counter.
96EE: A9 28   >127 :retry  lda    #r_GETMSG  ; Send GETMSG request.
96F0: 20 42 96 >128          jsr    REQUEST
96F3: B0 1C   >129          bcs    :failed    ; Timeout or no msg.
96F5: 20 F3 98 >130          jsr    la=>a      ; Set up address
>131          movl6 rbuf+len,length ; and length.
96F8: AD 4B 91 >131          lda    rbuf+len  ; Move 2 bytes
96FB: 85 FE   >131          sta    length
96FD: AD 4C 91 >131          lda    1+rbuf+len
9700: 85 FF   >131          sta    1+length
>131          eom
9702: 20 CF 99 >132          jsr    RCVLONG   ; Receive segmented message
9705: B0 0C   >133          bcs    :err      ; Timeout or cksum err.
>134          delay 40  ; Kill some time...
9707: A2 08   >134          ldx    #40/5     ; (5 cycles per iteration)
9709: CA     >134          ]delay
970A: D0 FD   >134          bne    ]delay
>134          eom
970C: A9 03   >135          lda    #rm_DACK ; -OK, send DATA ACK.
970E: 4C 14 98 >136          jmp    SENDRSP   ; and return w/ C clear.
>137

```

```
9711: D0 05      >138 :failed bne      :nak          ; Server has no message.
9713: CE 54 91  >139 :err   dec    reqretry ; Cksum or timeout; dec count.
9716: D0 D6      >140      bne      :retry       ; Try until OK or exhausted,
9718: 38          >141 :nak    sec          ; then return with C set.
9719: 60          >142      rts
```

```

60          put  SENDRCV
>1 *****
>2 *
>3 *          LOW-LEVEL PACKET FORMAT
>4 *          Revised ST Jun 27, 2005
>5 *
>6 * Start of packet:
>7 *
>8 *  --//---+---//---+          +-----+          +-----+-----+//-->
>9 *  Locked | ONE | ZERO | ONE | ZERO | ONE | Bit7 |
>10 *  or Idle| 31cy| 16cy | 8cy | 8cy | 8cy | 8cy |
>11 *  --//---+          +-----+          +-----+          +-----+//-->
>12 *          |          |          |          |          |
>13 *          |          | Start  | Coarse  | Servo   | <- 8 -//-->
>14 *          |          | sync   | sync   |         | data
>15 *          |          |         |         |         | bits
>16 *          |          |         |         |         | (64cy)
>17 *          |<----- Start sequence (71cy) ----->|
>18 *
>19 * (Note: data bits are transmitted inverted - 0-bit
>20 *       in memory is ONE on wire and vice versa)
>21 *
>22 * Interbyte separator:
>23 *
>24 *  >-//+-----+-----+          +-----+-----+-----+//-->
>25 *          |Bit1|Bit0|     ZERO     | ONE | Bit7|Bit6|
>26 *          |8cy |8cy | 22-23cy    | 8cy | 8cy | 8cy |
>27 *  >-//+-----+-----+          +-----+-----+//-->
>28 *          |          |          |          |          |
>29 *  >-//- 8 data -> |          | Servo   | <- 8 data -//-->
>30 *          bits   |          |         | bits
>31 *          |          |<----- Interbyte ----->|
>32 *          |          | separator
>33 *          |          | (30-31cy)
>34 *
>35 * Packet end:
>36 *
>37 *  >-//+-----+-----+
>38 *          |Bit1|Bit0|     ZERO (Idle)
>39 *          |8cy |8cy |
>40 *  >-//+-----+-----+//-->
>41 *          |
>42 *  >-//- End of -> |
>43 *          checkbyte
>44 *
>45 *****

```

```

>48 *****
>49 *
>50 *           B C A S T A R B
>51 *
>52 *           Michael J. Mahon - Aug 20, 2008
>53 *
>54 *           Copyright (c) 2008
>55 *
>56 * Broadcast Arbitrate is the precursor to any broadcast
>57 * request. Since there are no ACKs from receivers, it
>58 * takes steps to ensure that it controls the network
>59 * and all receivers are ready to receive data:
>60 *
>61 * 1. Arbitrate for and lock the network
>62 * 2. Delay 20ms. for any collisions to resolve and for
>63 *    any slow pollers to reach their RCVPKT holds
>64 * 3. Set 'sbuf+dst' to 0 for broadcast
>65 *
>66 *****
>67
971A: 20 00 98 >68 BCASTARB jsr   ARBTRATE   ; Arbitrate and lock network.
>69           dlyms 20       ; Let collisions resolve.
971D: A0 14   >69           ldy   #20       ; Delay 1ms. per iteration
>69           ldly  delay 1020-4 ; Cycles per ms. - 4
971F: A2 CB   >69           ldx   #1020-4/5 ; (5 cycles per iteration)
9721: CA     >69           ldelay dex
9722: D0 FD   >69           bne   ldelay
>69           eom
9724: 88     >69           dey
9725: D0 F8   >69           bne   ldly
>69           eom
9727: A9 00   >70           lda   #0         ; Set broadcast
9729: 8D 3F 91 >71           sta   sbuf+dst ; request.
972C: 60     >72           rts         ; and return.

```

```

>76 ]end align 256 ; Align to next page.
972D: 00 00 00 >76 ds *-1/256*256+256-*
>76 eom
>77 xmain equ *-]end ; (Timing-critical code)
>78
>79 *****
>80 *
>81 * A R B T R A T E *
>82 *
>83 * Michael J. Mahon - May 1, 1996 *
>84 * Revised Nov 05, 2004 *
>85 *
>86 * Copyright (c) 1996 *
>87 *
>88 * Waits until bus has been idle for 'arbtime' plus *
>89 * machine id # * 22 cycles, then locks bus and sends *
>90 * the request control packet. *
>91 *
>92 *****
>93
9800: AE 51 91 >94 ARBTRATE ldx arbxv ; Set arbitration wait.
9803: CD E8 C0 >95 cmp zipslow ; Zip Chip to 1MHz mode.
9806: 2C 62 C0 >96 :waitidl bit drecv ; Wait for idle bus.
9809: 30 F5 >97 bmi ARBTRATE ; Restart timing.
980B: CA >98 dex
980C: D0 F8 >99 bne :waitidl ; ...not yet.
980E: 8D 5B C0 >100 sta dsend+1 ; Got it! Lock the bus
9811: 60 >101 rts ; and return.

```

```

>103 *****
>104 *
>105 *           S E N D P K T
>106 *
>107 *           Michael J. Mahon - April 15, 1996
>108 *           Stephen Thomas - June 27, 2005
>109 *
>110 *           Copyright (c) 1996, 2003, 2004, 2005
>111 *
>112 * Sends (X) bytes (1..256) starting at (A,Y) to the
>113 * currently selected machine(s).
>114 *
>115 * SENDPKT does the following steps:
>116 *     1. Put Zip Chip in 'slow mode' for >38,000 cycles
>117 *     2. Send start signal: 31 cyc ONE, 16 cyc ZERO,
>118 *        8 cyc ONE, 8 cyc ZERO
>119 *     3. Send (X) data bytes (at 94-95 cyc/byte)
>120 *     4. Send one check byte (95 cyc), leaves bus ZERO
>121 *     5. Returns with Carry clear.
>122 *
>123 * SENDCTL performs a SENDPKT on the control packet
>124 * send buffer 'sbuf'.
>125 *
>126 * SENDRSP builds a packet specified by A in 'sbuf'
>127 * for the request in 'rbuf', then sends it.
>128 *
>129 * SENDACK builds an ACK packet in 'sbuf' for the
>130 * request in 'rbuf', then sends it.
>131 *
>132 * To obtain maximum sending speed (8 cycles/bit), the
>133 * inner loop of the actual sending code is unrolled
>134 * into a lattice, with two alternative straight-line
>135 * execution paths. One of these sends an alternating
>136 * sequence of ones and zeroes; the other sends the
>137 * inverse alternating sequence. Execution is bounced
>138 * from one path to the other depending on the data
>139 * being sent. Branch-taken delays are compensated for
>140 * by the fact that branches are only necessary when no
>141 * change in bus state is required.
>142 *
>143 *****
>144

```

```

9812: A9 02   >145 SENDACK  lda    #rm_ACK    ; Build an ACK packet
9814: 85 EC   >146 SENDRSP  sta    ckbyte   ; Store modifier for ora
9816: AD 45 91 >147         lda    rbuf+rqmd ; Get received request
9819: 29 F8   >148         and    #reqmask  ; isolate request
981B: 05 EC   >149         ora    ckbyte   ; and OR in modifier.
981D: 8D 3D 91 >150         sta    sbuf+rqmd ; Set response code.
9820: AD 48 91 >151         lda    rbuf+frm
9823: 8D 3F 91 >152         sta    sbuf+dst  ; Destination (= requester)
9826: A9 3D   >153 SENDCTL  lda    #<sbuf    ; Control pkt send buffer
9828: A0 91   >154         ldy    #>sbuf

```

```

982A: A2 08      >155          ldx    #lenctl
          >156
982C: CD E8 C0  >158  SENDPKT  cmp    zipslow    ; Slow Zip Chip for packet.
982F: 8D 5B C0  >162          sta    dsend+1    ; Send start signal ONE
9832: 20 9B 98  >163          jsr    :exit      ; Stretch it.
9835: 86 EC      >164          stx    ckbyte     ; Seed ckbyte with length.
9837: 84 EE      >165          sty    ptr+1     ; Y = start address hi
9839: A0 00      >166          ldy    #0        ; Index first data byte
983B: 18         >167          clc           ; Ensure C clear at exit
983C: 08         >168          php           ; Save interrupt state
983D: 78         >169          sei           ; and disable interrupts.
          >170
          >171  * Time-critical region. Timings for :tnn labels and
          >172  * t= comments are relative to the preceding timing point
          >173  * (start sync or servo).
          >174
983E: 8D 5A C0  >175          sta    dsend+0    ; Send start sync ZERO
9841: 2C 9B 98  >176          bit    :exit      ; Set V to send ckbyte at end
9844: 85 ED      >177          sta    ptr        ; A = start address lo
9846: B1 ED      >178          lda    (ptr),y    ; Get first data (Y=0 so no px)
9848: 8D 5B C0  >179          sta    dsend+1    ; Send coarse sync at t=16
984B: EA         >180          nop
984C: 38         >181          sec           ; Ensure C set between bytes
984D: 99 5A C0  >182          sta    dsend+0,y  ; Release coarse sync at t=24
9850: B0 03      >183          bcs    :servo     ; Go send servo at t=32
          >184
9852: C8         >185  :t84v0    iny           ; Get next data byte and
9853: B1 ED      >186          lda    (ptr),y    ; send servo at t=94 or 95
          >187
9855: 8D 5B C0  >188  :servo    sta    dsend+1    ; Servo ONE
9858: 2A         >189          rol
9859: 90 41      >190          bcc    :t06b7v1
985B: 8D 5A C0  >191          sta    dsend+0    ; Bit 7 ZERO at t=8
985E: 2A         >192          rol
985F: B0 3D      >193          bcs    :t14b6v0
9861: 8D 5B C0  >194          sta    dsend+1    ; Bit 6 ONE at t=16
9864: 2A         >195  :t17b6v1  rol
9865: 90 39      >196          bcc    :t22b5v1
9867: 8D 5A C0  >197          sta    dsend+0    ; Bit 5 ZERO at t=24
986A: 2A         >198  :t25b5v0  rol
986B: B0 35      >199          bcs    :t30b4v0
986D: 8D 5B C0  >200          sta    dsend+1    ; Bit 4 ONE at t=32
9870: 2A         >201  :t33b4v1  rol
9871: 90 31      >202          bcc    :t38b3v1
9873: 8D 5A C0  >203          sta    dsend+0    ; Bit 3 ZERO at t=40
9876: 2A         >204  :t41b3v0  rol
9877: B0 2D      >205          bcs    :t46b2v0
9879: 8D 5B C0  >206          sta    dsend+1    ; Bit 2 ONE at t=48
987C: 2A         >207  :t49b2v1  rol
987D: 90 29      >208          bcc    :t54b1v1
987F: 8D 5A C0  >209          sta    dsend+0    ; Bit 1 ZERO at t=56
9882: 2A         >210  :t57b1v0  rol

```



```

9883: B0 25 >211 bcs :t62b0v0
9885: 8D 5B C0 >212 sta dsend+1 ; Bit 0 ONE at t=64
9888: 2A >213 :t65b0v1 rol ; Restore data, set C
9889: EA >214 nop
988A: 8D 5A C0 >215 sta dsend+0 ; Idle/interbyte ZERO at t=72
>216
988D: 45 EC >217 :t73v0 eor ckbyte ; Compute checksum
988F: 85 EC >218 sta ckbyte ; and save it.
9891: CA >219 dex ; Count bytes sent
9892: D0 BE >220 bne :t84v0 ; Loop while more to send
>221
9894: 50 04 >222 :t83v0 bvc :done ; Quit if ckbyte already sent;
9896: E8 >223 inx ; else count ckbyte,
9897: B8 >224 clv ; clear send-ckbyte flag,
9898: 50 BB >225 bvc :servo ; Send ckbyte servo at t=95
>226
989A: 28 >227 :done plp ; Restore int state
989B: 60 >228 :exit rts ; and return with C clear.
>229
989C: 90 1E >230 :t06b7v1 bcc :t09b7v1 ; These are all for timing
989E: B0 22 >231 :t14b6v0 bcs :t17b6v0 ; equalization and all of
98A0: 90 26 >232 :t22b5v1 bcc :t25b5v1 ; them are always taken
98A2: B0 2A >233 :t30b4v0 bcs :t33b4v0
98A4: 90 2E >234 :t38b3v1 bcc :t41b3v1
98A6: B0 32 >235 :t46b2v0 bcs :t49b2v0
98A8: 90 36 >236 :t54b1v1 bcc :t57b1v1
98AA: B0 3A >237 :t62b0v0 bcs :t65b0v0
>238
98AC: 90 B6 >239 :t14b6v1 bcc :t17b6v1
98AE: B0 BA >240 :t22b5v0 bcs :t25b5v0
98B0: 90 BE >241 :t30b4v1 bcc :t33b4v1
98B2: B0 C2 >242 :t38b3v0 bcs :t41b3v0
98B4: 90 C6 >243 :t46b2v1 bcc :t49b2v1
98B6: B0 CA >244 :t54b1v0 bcs :t57b1v0
98B8: 90 CE >245 :t62b0v1 bcc :t65b0v1
98BA: B0 D1 >246 :t70v0 bcs :t73v0
>247
98BC: 2A >248 :t09b7v1 rol
98BD: 90 ED >249 bcc :t14b6v1
98BF: 8D 5A C0 >250 sta dsend+0 ; Bit 6 ZERO at t=16
98C2: 2A >251 :t17b6v0 rol
98C3: B0 E9 >252 bcs :t22b5v0
98C5: 8D 5B C0 >253 sta dsend+1 ; Bit 5 ONE at t=24
98C8: 2A >254 :t25b5v1 rol
98C9: 90 E5 >255 bcc :t30b4v1
98CB: 8D 5A C0 >256 sta dsend+0 ; Bit 4 ZERO at t=32
98CE: 2A >257 :t33b4v0 rol
98CF: B0 E1 >258 bcs :t38b3v0
98D1: 8D 5B C0 >259 sta dsend+1 ; Bit 3 ONE at t=40
98D4: 2A >260 :t41b3v1 rol
98D5: 90 DD >261 bcc :t46b2v1
98D7: 8D 5A C0 >262 sta dsend+0 ; Bit 2 ZERO at t=48

```

```
98DA: 2A      >263  :t49b2v0 rol
98DB: B0 D9   >264           bcs   :t54b1v0
98DD: 8D 5B C0 >265           sta   dsend+1   ; Bit 1 ONE at t=56
98E0: 2A      >266  :t57b1v1 rol
98E1: 90 D5   >267           bcc   :t62b0v1
98E3: 8D 5A C0 >268           sta   dsend+0   ; Bit 0 ZERO at t=64
98E6: 2A      >269  :t65b0v0 rol   ; Restore data, set C
98E7: B0 D1   >270           bcs   :t70v0     ; Always taken
```

```

>272 *****
>273 *
>274 *           L A S L = > A L
>275 *
>276 *           L A = > A
>277 *
>278 *****

```

```

>279
>280 lasl=>al mov16 sbuf+len;length ; 'sbuf' length -> length

```

```

98E9: AD 43 91 >280      lda  sbuf+len ; Move 2 bytes

```

```

98EC: 85 FE      >280      sta  length

```

```

98EE: AD 44 91 >280      lda  1+sbuf+len

```

```

98F1: 85 FF      >280      sta  1+length

```

```

>280      eom

```

```

>281 la=>a      mov16 locaddr;address ; Local address -> address

```

```

98F3: AD 4D 91 >281      lda  locaddr ; Move 2 bytes

```

```

98F6: 85 FC      >281      sta  address

```

```

98F8: AD 4E 91 >281      lda  1+locaddr

```

```

98FB: 85 FD      >281      sta  1+address

```

```

>281      eom

```

```

98FD: 60        >282      rts

```

98FE: 00 00

```

>284 ]end      align 256          ; Align to next page.
>284          ds      *-1/256*256+256-*
>284          eom
>285 xsend     equ      *-]end      ; (Timing-critical code)
>287
>288 *****
>289 *
>290 *                      R C V P K T
>291 *
>292 *                      Michael J. Mahon - April 15, 1996
>293 *                      Stephen Thomas - June 27, 2005
>294 *                      Revised May 21, 2008
>295 *
>296 *                      Copyright (c) 1996, 2003, 2004, 2005, 2008
>297 *
>298 *  Receives (X) bytes (1..256) starting at (A,Y) from
>299 *  the sending machine.
>300 *
>301 *  If no packet is detected within the minimum arb time
>302 *  plus 'tolim'-1 times 2.8ms, it returns with carry set
>303 *  and A = 0.
>304 *
>305 *  If packet is received, but checksum doesn't compare,
>306 *  it returns with carry set and A <> 0.
>307 *
>308 *  RCVPKT does the following steps:
>309 *      1. Detect 'start signal' ONE
>310 *      2. Put Zip Chip in 'slow mode' for >38,000 cycles
>311 *      3. Sync to cycles 5-7 of 8-cycle data cells
>312 *      3. Receive (X) bytes (at 93 +3/-0 cycles/byte)
>313 *      4. Receive check byte and verify correctness,
>314 *          keeping count of checksum errors.
>315 *
>316 *  RCVCTL performs a RCVPKT to the control packet
>317 *  receive buffer 'rbuf'.
>318 *
>319 *  RCVPTR performs a RCVPKT to the address in 'ptr' with
>320 *  length (X).
>321 *
>322 *****
>323 *
>324 *                      Implementation Note
>325 *
>326 *  RCVPKT maintains synchronization with the data stream
>327 *  by using a "digital PLL" technique.  The RCVPKT byte
>328 *  loop is 93 cycles, which is 1 or 2 cycles shorter
>329 *  than the send loop.  When RCVPKT samples the servo
>330 *  transition and finds that it hasn't happened yet, it
>331 *  adds a 3-cycle delay to make the total loop time 96
>332 *  cycles and restore optimal sync.
>333 *
>334 *  The effect is to keep the data sampling window on the

```

```

>335 * 5th to 7th cycle of the 8-cycle data bitcell, in *
>336 * spite of the send loop buffer crossing pages at some *
>337 * point in a packet and clock frequency differences of *
>338 * +/- 1% between sending and receiving machines. *
>339 * *
>340 * A similar technique assures a well-controlled sample *
>341 * position from the first byte of each received packet: *
>342 * *
>343 * After the ONE marking the packet start, there's a 16 *
>344 * cycle ZERO. Call the time the transmitter begins *
>345 * that ZERO t=0. *
>346 * *
>347 * The receive loop waits for the ZERO, sampling the *
>348 * bus in a tight loop with a 7-cycle period; call the *
>349 * time its first ZERO sample occurs rt=0. Allowing up *
>350 * to 4 cycles for pulldown time on the worst network *
>351 * bus we can possibly work with, rt=0 could be any time *
>352 * between t=0 and t=11. *
>353 * *
>354 * At t=16, the transmitter will actively drive the bus *
>355 * to ONE (a hard-driven transition typically taking *
>356 * much less than 1 cycle). At rt=10, the receive code *
>357 * samples the bus once again; if it sees ONE (which it *
>358 * will only do if rt=0 occurred between t=6 and t=11) *
>359 * it skips a 6-cycle time delay, arriving at rt=19 six *
>360 * cycles early. This makes the rest of the timing work *
>361 * as if rt=0 had actually fallen between t=0 and t=5 *
>362 * instead of t=6 and t=11. Timings referred to rt=0 *
>363 * now have an uncertainty of only 6 cycles with respect *
>364 * to t=0 instead of the 11 cycle uncertainty they began *
>365 * with, and the receiver is in coarse sync. *
>366 * *
>367 * In the most-delayed case, with rt=0 at t=11, the *
>368 * rt=10 sample will occur at t=21. Since the trans- *
>369 * mitter does not release the bus until t=24, this is *
>370 * safe. *
>371 * *
>372 * At t=32, the transmitter will drive the bus back to *
>373 * ONE. At rt=29, the receive code samples the bus and *
>374 * if it sees ONE (which it will only do if rt=0 fell *
>375 * between t=3 and t=6) it skips a 3-cycle time delay, *
>376 * arriving at rt=36 three cycles early. This makes the *
>377 * rest of the timing work as if rt=0 actually happened *
>378 * between t=0 and t=3 instead of t=3 and t=6. Timings *
>379 * referred to rt=0 now have an uncertainty of only 3 *
>380 * cycles with respect to t=0, and the receiver is in *
>381 * fine sync. *
>382 * *
>383 * The edge at t=32 is actually the servo edge for the *
>384 * first byte. Timings within a data byte are all taken *
>385 * relative to the servo edge, so t=32 is redefined as *
>386 * t=0 and a corresponding adjustment is made to rt; *

```

```

>387 * the point called rt=36 in the previous paragraph is *
>388 * actually labelled :rt04 in the code. *
>389 * *
>390 * The first data bitcell runs from t=8 to t=16. The *
>391 * receiver samples it at rt=12 - that is, some time *
>392 * between t=12 and t=15. This gives a 4-cycle margin *
>393 * at the start of the bitcell and 1 cycle at the end, *
>394 * which should be reliable even with truly woeful *
>395 * pulldown times. *
>396 * *
>397 * Samples for the rest of the data bits are taken at *
>398 * 8-cycle intervals to match the transmit rate, and the *
>399 * 3-cycle fine sync code is re-used to implement the *
>400 * DPLL and make sure the receiver stays in sync for all *
>401 * subsequent data bytes. *
>402 * *
>403 *****
>404

```

```

9900: A9 45 >405 RCVCTL lda #<rbuf ; Receive control pkt to 'rbuf'
9902: A0 91 >406 ldy #>rbuf
9904: A2 08 >407 ldx #lenctl
9906: 85 ED >408 RCVPKT sta ptr ; A = buf address lo
9908: 84 EE >409 sty ptr+1 ; Y = buf address hi
990A: 8A >410 RCVPTR txa ; Seed checksum with length
990B: CA >411 dex ; X = length 1..256 (0=>256);
990C: 86 EB >412 stx lastidx ; convert to last buffer index
>413
990E: AC 52 91 >414 ldy tolim ; Wait <= (tolim-1) * 2.8ms.
9911: A2 5C >415 ldx #arbx ; plus minimum arb time.
9913: 08 >416 php ; Save interrupt state
9914: 78 >417 sei ; and disable interrupts.
9915: 2C E8 C0 >418 bit zipslow ; Slow any Zip Chip to 1 MHz.
>419
9918: 2C 62 C0 >420 :waitstr bit drcv ; Wait for starting ONE.
991B: 30 0A >421 bmi :gotstr
991D: CA >422 dex ; (inner loop is 11 cycles)
991E: D0 F8 >423 bne :waitstr ; Keep waiting...
9920: 88 >424 dey ; (outer loop is 2820 cycles)
9921: D0 F5 >425 bne :waitstr ; Loop for 'timeout' ms.
>426
9923: 28 >427 plp ; Restore int state
9924: 98 >428 tya ; Signal timeout (A=0, Z set)
9925: 38 >429 sec ; and return with C set.
9926: 60 >430 :exit rts
>431
9927: 2C E8 C0 >432 :gotstr bit zipslow ; Slow Zip Chip for packet.
>433
992A: 2C 62 C0 >434 :waitsyn bit drcv ; Wait for 16-cycle sync ZERO;
992D: 30 FB >435 bmi :waitsyn ; too bad if bus locks forever!
>436
992F: A0 FF >437 ldy #$FF ; Index-1 of first data location
9931: A2 7F >438 ldx #$7F ; CPX #0-7F sets C, 80-FF clears

```

```

9933: EC 62 C0 >439 :synrt07 cpx drecv ; Check for coarse sync at rt=10
9936: B0 05 >440 bcs :synrt14 ; Only do delay if still ZERO
>441
9938: 2C 26 99 >442 :synrt19 bit :exit ; Set V (not-ckbyte flag)
993B: 70 04 >443 bvs :servo ; Do first servo check at rt=29
>444
993D: 18 >445 :synrt14 clc ; 6-cycle coarse sync delay
993E: 90 F8 >446 bcc :synrt19 ; (1 extra to get here, 5 back)
>447
9940: B8 >448 :rt88 clv ; Clear not-ckbyte flag
>449
9941: EC 62 C0 >450 :servo cpx drecv ; Check for servo transition
9944: 90 02 >451 :rt01 bcc :rt04 ; Delay 3 cyc if past servo,
9946: EA >452 nop ; 6 if not
9947: EA >453 nop
>454
9948: 85 EC >455 :rt04 sta ckbyte ; Update checksum
994A: C8 >456 iny ; Index next data location
>457
994B: EC 62 C0 >458 :rt09 cpx drecv ; C <-- ~ bit 7 at rt=12
994E: 2A >459 rol ; Shift bit 7 in.
994F: EA >460 nop
9950: EC 62 C0 >461 cpx drecv ; C <-- ~ bit 6 at rt=20
9953: 2A >462 rol
9954: EA >463 nop
9955: EC 62 C0 >464 cpx drecv ; C <-- ~ bit 5 at rt=28
9958: 2A >465 rol
9959: EA >466 nop
995A: EC 62 C0 >467 cpx drecv ; C <-- ~ bit 4 at rt=36
995D: 2A >468 rol
995E: EA >469 nop
995F: EC 62 C0 >470 cpx drecv ; C <-- ~ bit 3 at rt=44
9962: 2A >471 rol
9963: EA >472 nop
9964: EC 62 C0 >473 cpx drecv ; C <-- ~ bit 2 at rt=52
9967: 2A >474 rol
9968: EA >475 nop
9969: EC 62 C0 >476 cpx drecv ; C <-- ~ bit 1 at rt=60
996C: 2A >477 rol
996D: EA >478 nop
996E: EC 62 C0 >479 cpx drecv ; C <-- ~ bit 0 at rt=68
9971: 2A >480 :rt69 rol
9972: 50 0A >481 :rt71 bvc :rcvdone ; quit after ckbyte
>482
9974: 91 ED >483 :rt73 sta (ptr),y ; Save data (always 6cy)
9976: 45 EC >484 :rt79 eor ckbyte ; Compute checksum
9978: C4 EB >485 :rt82 cpy lastidx ; Stored last byte?
997A: F0 C4 >486 :rt85 beq :rt88 ; Go clear not-ckbyte flag if so
997C: D0 C3 >487 :rt87 bne :servo ; Do next servo sample at rt=93
>488
997E: 45 EC >489 :rcvdone eor ckbyte ; A = 0 if ckbyte = sum
9980: F0 08 >490 beq :goodck ; -No error.

```

```

          >491          incl6 ckerr      ; Count checksum error.
9982: EE 58 91 >491    inc      ckerr      ; Increment 16-bit word.
9985: D0 03          >491    bne      *+5        ; - No carry.
9987: EE 59 91 >491    inc      ckerr+1     ; Propagate carry.
          >491          eom
998A: 28            >492    :goodck plp          ; Restore int state
998B: C9 01          >493    cmp      #1         ; Set C & NZ if checksum bad,
998D: AA            >494    tax          ; clear C and set Z if good
998E: 60            >495    rts          ; and return.
```



```

>497 *****
>498 *
>499 *           P R O T E R R
>500 *
>501 *****
>502
>503 PROTERR  incl6  errprot    ; Count protocol error.
998F: EE 56 91 >503          inc    errprot    ; Increment 16-bit word.
9992: D0 03    >503          bne    *+5      ; - No carry.
9994: EE 57 91 >503          inc    errprot+1 ; Propagate carry.
>503          eom
9997: 60      >504          rts
>505
>506
>507 *****
>508 *
>509 *           R A R L = > A L
>510 *
>511 *           R A = > A
>512 *
>513 *****
>514
>515 rarl=>a1 mov16 rbuf+len;length ; 'rbuf' length -> length
9998: AD 4B 91 >515          lda    rbuf+len  ; Move 2 bytes
999B: 85 FE    >515          sta    length
999D: AD 4C 91 >515          lda    1+rbuf+len
99A0: 85 FF    >515          sta    1+length
>515          eom
>516 ra=>a    mov16 rbuf+adr;address; 'rbuf' address -> address
99A2: AD 49 91 >516          lda    rbuf+adr  ; Move 2 bytes
99A5: 85 FC    >516          sta    address
99A7: AD 4A 91 >516          lda    1+rbuf+adr
99AA: 85 FD    >516          sta    1+address
>516          eom
99AC: 60      >517          rts

```

```

>520 *****
>521 *
>522 *           S E N D L O N G
>523 *
>524 *           Michael J. Mahon - May 5, 1996
>525 *           Revised May 21, 2008
>526 *
>527 *           Copyright (c) 1996, 2008
>528 *
>529 * SENDLONG sends 'length' bytes from 'address' to the
>530 * currently selected machine(s).
>531 *
>532 * DSENLNG delays X*5-1 cycles and falls into SENDLONG.
>533 *
>534 * It segments a "message" longer than 256 bytes into a
>535 * series of 256-byte packets, plus a final packet
>536 * with the remainder of the data. Each message packet
>537 * is sent with 'SENDPKT'.
>538 *
>539 * SENDLONG does not detect any errors.
>540 *
>541 *****
>542

```

```

99AD: CA      >543 DSENLNG dex          ; Delay 5 * X - 1 cycles
99AE: D0 FD   >544         bne DSENLNG ; and fall into SENDLONG.
99B0: A5 FF   >545 SENDLONG lda length+1 ; How many 256-byte pages?
99B2: F0 0F   >546         beq :short   ; - None, just a short pkt.
99B4: A2 00   >547 :loop   ldx #0        ; Set 256 byte packet.
99B6: A5 FC   >548         lda address  ; and point to
99B8: A4 FD   >549         ldy address+1 ; data buffer.
99BA: 20 2C 98 >550         jsr SENDPKT  ; Send 256 bytes.
99BD: E6 FD   >551         inc address+1 ; Advance to next page
99BF: C6 FF   >552         dec length+1 ; and decrement page
99C1: D0 F1   >553         bne :loop   ; count until done...
99C3: A6 FE   >554 :short  ldx length   ; Remaining data length.
99C5: F0 07   >555         beq :done   ; -All done.
99C7: A5 FC   >556         lda address
99C9: A4 FD   >557         ldy address+1
99CB: 20 2C 98 >558         jsr SENDPKT  ; Send the final packet.
99CE: 60      >559 :done   rts

```

```

>562 *****
>563 *
>564 *           R C V L O N G
>565 *
>566 *           Michael J. Mahon - May 5, 1996
>567 *           Revised May 21, 2008
>568 *
>569 *           Copyright (c) 1996, 2008
>570 *
>571 * RCVLONG receives 'length' bytes to 'address' from the
>572 * currently sending machine.
>573 *
>574 * It receives a series of packets if 'length' is
>575 * greater than 256 bytes.
>576 *
>577 * RCVLONG detects checksum errors and timeouts, and
>578 * returns with Carry set and A=0 if timeout, and
>579 * Carry set and A>0 if a checksum error. Timeouts in
>580 * this context are protocol errors. Both kinds of
>581 * errors are tallied in counters.
>582 *
>583 *****
>584

```

```

99CF: A5 FF >585 RCVLONG lda length+1 ; How many 256-byte pages?
99D1: F0 11 >586 beq :short ; - None, just a short pkt.
99D3: A2 00 >587 :loop ldx #0 ; Set 256 byte packet.
99D5: A5 FC >588 lda address ; and point to
99D7: A4 FD >589 ldy address+1 ; data buffer.
99D9: 20 06 99 >590 jsr RCVPKT ; Receive 256 bytes.
99DC: B0 14 >591 bcs :err ; Receive error detected.
99DE: E6 FD >592 inc address+1 ; Advance to next page
99E0: C6 FF >593 dec length+1 ; and decrement page
99E2: D0 EF >594 bne :loop ; count until done...
99E4: A6 FE >595 :short ldx length ; Remaining data length.
99E6: F0 09 >596 beq :done ; -All done.
99E8: A5 FC >597 lda address
99EA: A4 FD >598 ldy address+1
99EC: 20 06 99 >599 jsr RCVPKT ; Receive final packet.
99EF: B0 01 >600 bcs :err ; Keep track of any errors.
99F1: 60 >601 :done rts
>602
99F2: D0 03 >603 :err bne :ckerr ; Checksum error.
99F4: 20 8F 99 >604 jsr PROTERR ; Count protocol error.
99F7: A8 >605 :ckerr tay ; Set Z flag from A.
99F8: 60 >606 rts

```

```

61      ]end      align 256      ; Align to page boundary
99F9: 00 00 00 61      ds      *-1/256*256+256-*
61      eom
62      xreceive equ      *-]end      ; Extra space at end.
63      err      *-1-entry/SIZE ; Can't exceed limit
    
```

--End assembly, 2304 bytes, Errors: 0

Symbol table - alphabetical order:

@	=\$913B	ADDON	=\$D998	AMPNADA	=\$9254	AMPVECT	=\$03F5
ARBTRATE	=\$9800	AX	=\$48	BCASTARB	=\$971A	BCASTREQ	=\$94A1
? BELL	=\$FF3A	BOOTREQ	=\$949D	BPOKEREQ	=\$9617	BPOKESRV	=\$9622
BRUNREQ	=\$957B	BRUNSRV	=\$95BC	CALLREQ	=\$9631	CALLSRV	=\$9636
CALL_t	=\$8C	CHRGET	=\$B1	CHRGOT	=\$B7	COLDSTRT	=\$E000
COUT	=\$FDED	CROUT1	=\$FD8B	CSW	=\$36	DSNDLNG	=\$99AD
ERROR	=\$D412	FAC	=\$9D	FIXLINKS	=\$D4F2	FLO2	=\$EBA0
FORPNT	=\$85	FRETOP	=\$6F	FRMNUM	=\$DD67	GETADR	=\$E752
GETBYT	=\$E6F8	GETIDSRV	=\$94B4	GETMREQ	=\$96E9	GET_t	=\$BE
HIMEM	=\$73	? HOME	=\$FC58	INIT	=\$93B7	INSTALL	=\$9229
INTFLG	=\$12	KSW	=\$38	ONERR	=\$D8	PEEKREQ	=\$94E3
PEEKSRV	=\$9514	PEEK_t	=\$E2	PKINCREQ	=\$9537	PKINCSRV	=\$9554
PKPOKREQ	=\$953B	PKPOKSRV	=\$9558	POKEREQ	=\$957F	POKESRV	=\$95BC
POKE_t	=\$B9	? PRBL2	=\$F94A	PRBYTE	=\$FDDA	? PREAD	=\$FB1E
? PROGEND	=\$AF	PROTERR	=\$998F	PSTART	=\$67	PTRGET	=\$DFE3
PUTMREQ	=\$96C7	? PWREDUP	=\$03F4	? RARL=>AL	=\$9136	RCVCTL	=\$9900
RCVDACK	=\$969E	RCVLONG	=\$99CF	RCVPKT	=\$9906	RCVPTR	=\$990A
REQUEST	=\$9642	ROMboot	=\$00	RUNPROG	=\$D566	RUNREQ	=\$9577
RUNSRV	=\$95A7	RUN_t	=\$AC	SENDACK	=\$9812	SENDCTL	=\$9826
SENDLONG	=\$99B0	SENDPKT	=\$982C	SENDRSP	=\$9814	SERVER	=\$9406
SETFOR	=\$EB27	SIMPLREQ	=\$953D	SIZE	=\$0900	? SOFTEV	=\$03F2
SYNCHR	=\$DEC0	SYNERR	=\$DEC9	TXTPTR	=\$B8	VALTYP	=\$11
VARTAB	=\$69	? VBL	=\$C019	V? ]PROTERR	=\$9471	V ]cpx	=\$0B
V ]cpy	=\$0B04	V ]cy	=\$4FB0	MV ]delay	=\$9721	MV ]dly	=\$971F
V? ]doit	=\$94A3	V ]end	=\$99F9	V ]servpad	=\$FF	addr	=\$46
address	=\$FC	adr	=\$04	MD align	=\$8000	an0	=\$C058
an1	=\$C05A	? an2	=\$C05C	? an3	=\$C05E	arbtime	=\$01
arbx	=\$5C	arbxv	=\$9151	? bcast	=\$911E	bootself	=\$03CC
? bpoke	=\$9121	? brun	=\$912D	byte	=\$00	? call	=\$9115
chain	=\$926A	ckbyte	=\$EC	ckerr	=\$9158	class	=\$46
cmd	=\$9261	cmdptr	=\$EC	cmdsave	=\$ED	cmdtable	=\$9184
comp	=\$926D	crate	=\$00	cyperms	=\$03FC	MD delay	=\$8000
dest	=\$04	disp	=\$EF	MD dlyms	=\$8000	docall	=\$9639
dos	=\$00	drecv	=\$C062	dsend	=\$C05A	dsk6off	=\$C0E8
dst	=\$02	enhboot	=\$00	entry	=\$9100	errprot	=\$9156
errstop	=\$917F	frm	=\$03	frmc	=\$01	frmcerr	=\$915A
? gapwait	=\$07	? getmsg	=\$911B	idletime	=\$14	idleto	=\$08
idtable	=\$915D	idtbl	=\$937E	MD incl6	=\$8000	incr	=\$48
? init	=\$9109	instald	=\$917D	iter	=\$15	kbstrobe	=\$C010
keybd	=\$C000	la=>a	=\$98F3	lasl=>al	=\$98E9	lastidx	=\$EB

len	=\$06	lenctl	=\$08	length	=\$FE	lngth	=\$48
lngth?	=\$D0	locaddr	=\$914D	locadr	=\$52	master	=\$01
? maxarb	=\$03	maxgap	=\$57	maxid	=\$1F	maxreq	=\$70
maxreqrt	=\$03	maxretry	=\$32	modmask	=\$07	MD mov16	=\$8000
mserve	=\$00	n60ms	=\$14	nadapage	=\$03CF	? nadaver	=\$915C
nparms	=\$917E	null	=\$937C	parmsiz	=\$15	pb0	=\$C061
pb1	=\$C062	? pb2	=\$C063	? peek	=\$910F	? peekinc	=\$9124
? peekpoke	=\$9127	pkxxxxsrv	=\$955A	? poke	=\$9112	ptr	=\$ED
? ptrig	=\$C070	? putmsg	=\$9118	r_BCAST	=\$40	r_BOOT	=\$38
r_BPOKE	=\$48	r_BRUN	=\$68	r_CALL	=\$18	? r_GETID	=\$30
r_GETMSG	=\$28	r_PEEK	=\$08	r_PKINC	=\$50	r_PKPOK	=\$58
r_POKE	=\$10	r_PUTMSG	=\$20	r_RUN	=\$60	ra=>a	=\$99A2
rarl=>al	=\$9998	rbuf	=\$9145	? rcvctl	=\$9130	? rcvlong	=\$9139
? rcvptr	=\$9133	reqctr	=\$9153	reqdelay	=\$11	reqdur	=\$06
reqfac	=\$08	reqmask	=\$F8	reqpidle	=\$03	reqretry	=\$9154
reqtime	=\$0BB8	reqto	=\$01	retrycnt	=\$9155	retrylim	=\$914F
rm_ACK	=\$02	rm_DACK	=\$03	rm_NAK	=\$04	rm_REQ	=\$01
rqmd	=\$00	rqperiod	=\$14	rts	=\$9616	? run	=\$912A
savhooks	=\$02FC	sbuf	=\$913D	self	=\$913C	? serve	=\$910C
servecnt	=\$9150	servegap	=\$3A	servelp	=\$9103	service	=\$938A
sethooks	=\$95A3	setid	=\$93E9	setreq	=\$9581	? spkr	=\$C030
svrxkbd	=\$9403	? t_BASIC	=\$E0	? t_SYNTH	=\$F0	? t_VOICE	=\$F1
timeout	=\$9372	tolim	=\$9152	val	=\$48	val?	=\$D0
var	=\$80	varadr	=\$9182	varcmd	=\$9180	vartype	=\$9181
verlen	=\$13	vermsg	=\$93A4	version	=\$31	warmstrt	=\$03CD
word	=\$40	? xmain	=\$D3	? xreceive	=\$07	? xsend	=\$02
zipslow	=\$C0E8						

Symbol table - numerical order:

dos	=\$00	crate	=\$00	mserve	=\$00	ROMboot	=\$00
enhboot	=\$00	rqmd	=\$00	byte	=\$00	master	=\$01
aritime	=\$01	reqto	=\$01	frmc	=\$01	rm_REQ	=\$01
dst	=\$02	rm_ACK	=\$02	? xsend	=\$02	? maxarb	=\$03
reqpidle	=\$03	maxreqrt	=\$03	frm	=\$03	rm_DACK	=\$03
adr	=\$04	rm_NAK	=\$04	dest	=\$04	reqdur	=\$06
len	=\$06	? gapwait	=\$07	modmask	=\$07	? xreceive	=\$07
idleto	=\$08	lenctl	=\$08	reqfac	=\$08	r_PEEK	=\$08
V ]cpx	=\$0B	r_POKE	=\$10	reqdelay	=\$11	VALTYP	=\$11
INTFLG	=\$12	verlen	=\$13	idletime	=\$14	rqperiod	=\$14
n60ms	=\$14	parmsiz	=\$15	iter	=\$15	r_CALL	=\$18
maxid	=\$1F	r_PUTMSG	=\$20	r_GETMSG	=\$28	? r_GETID	=\$30
version	=\$31	maxretry	=\$32	CSW	=\$36	KSW	=\$38
r_BOOT	=\$38	servegap	=\$3A	r_BCAST	=\$40	word	=\$40
addr	=\$46	class	=\$46	r_BPOKE	=\$48	lngth	=\$48
AX	=\$48	incr	=\$48	val	=\$48	r_PKINC	=\$50
locadr	=\$52	maxgap	=\$57	r_PKPOK	=\$58	arbx	=\$5C
r_RUN	=\$60	PSTART	=\$67	r_BRUN	=\$68	VARTAB	=\$69
FRETOP	=\$6F	maxreq	=\$70	HIMEM	=\$73	var	=\$80
FORPNT	=\$85	CALL_t	=\$8C	FAC	=\$9D	RUN_t	=\$AC
? PROGEND	=\$AF	CHRGET	=\$B1	CHRGOT	=\$B7	TXTPTR	=\$B8

POKE_t	=\$B9	GET_t	=\$BE	lngth?	=\$D0	val?	=\$D0
? xmain	=\$D3	ONERR	=\$D8	? t_BASIC	=\$E0	PEEK_t	=\$E2
lastidx	=\$EB	ckbyte	=\$EC	cmdptr	=\$EC	ptr	=\$ED
cmdsave	=\$ED	disp	=\$EF	? t_SYNTH	=\$F0	? t_VOICE	=\$F1
reqmask	=\$F8	address	=\$FC	length	=\$FE	V ]servpad	=\$FF
savhooks	=\$02FC	bootself	=\$03CC	warmstrt	=\$03CD	nadapage	=\$03CF
? SOFTEV	=\$03F2	? PWREDUP	=\$03F4	AMPVECT	=\$03F5	cyperms	=\$03FC
SIZE	=\$0900	V ]cpy	=\$0B04	reqtime	=\$0BB8	V ]cy	=\$4FB0
entry	=\$9100	servelp	=\$9103	? init	=\$9109	? serve	=\$910C
? peek	=\$910F	? poke	=\$9112	? call	=\$9115	? putmsg	=\$9118
? getmsg	=\$911B	? bcast	=\$911E	? bpoke	=\$9121	? peekinc	=\$9124
? peekpoke	=\$9127	? run	=\$912A	? brun	=\$912D	? rcvctl	=\$9130
? rcvptr	=\$9133	? RARL=>AL	=\$9136	? rcvlong	=\$9139	@	=\$913B
self	=\$913C	sbuf	=\$913D	rbuf	=\$9145	locaddr	=\$914D
retrylim	=\$914F	servecnt	=\$9150	arbxv	=\$9151	tolim	=\$9152
reqctr	=\$9153	reqretry	=\$9154	retrycnt	=\$9155	errprot	=\$9156
ckerr	=\$9158	frmcerr	=\$915A	? nadaver	=\$915C	idtable	=\$915D
instald	=\$917D	nparms	=\$917E	errstop	=\$917F	varcmd	=\$9180
vartype	=\$9181	varadr	=\$9182	cmdtable	=\$9184	INSTALL	=\$9229
AMPNADA	=\$9254	cmd	=\$9261	chain	=\$926A	comp	=\$926D
timeout	=\$9372	null	=\$937C	idtbl	=\$937E	service	=\$938A
vermsg	=\$93A4	INIT	=\$93B7	setid	=\$93E9	svrxkbd	=\$9403
SERVER	=\$9406	V? ]PROTERR	=\$9471	BOOTREQ	=\$949D	BCASTREQ	=\$94A1
V? ]doit	=\$94A3	GETIDSRV	=\$94B4	PEEKREQ	=\$94E3	PEEKSRV	=\$9514
PKINCREQ	=\$9537	PKPOKREQ	=\$953B	SIMPLREQ	=\$953D	PKINCSRV	=\$9554
PKPOKSRV	=\$9558	pkxxsrv	=\$955A	RUNREQ	=\$9577	BRUNREQ	=\$957B
POKEREQ	=\$957F	setreq	=\$9581	sethooks	=\$95A3	RUNSRV	=\$95A7
BRUNSRV	=\$95BC	POKESRV	=\$95BC	rts	=\$9616	BPOKEREQ	=\$9617
BPOKESRV	=\$9622	CALLREQ	=\$9631	CALLSRV	=\$9636	docal	=\$9639
REQUEST	=\$9642	RCVDACK	=\$969E	PUTMREQ	=\$96C7	GETMREQ	=\$96E9
BCASTARB	=\$971A	MV ]dly	=\$971F	MV ]delay	=\$9721	ARBTRATE	=\$9800
SENDACK	=\$9812	SENDRSP	=\$9814	SENDCTL	=\$9826	SENDPKT	=\$982C
lasl=>al	=\$98E9	la=>a	=\$98F3	RCVCTL	=\$9900	RCVPKT	=\$9906
RCVPTR	=\$990A	PROTERR	=\$998F	rarl=>al	=\$9998	ra=>a	=\$99A2
DSENDLNG	=\$99AD	SENDRONG	=\$99B0	RCVLONG	=\$99CF	V ]lend	=\$99F9
MD align	=\$8000	MD dlyms	=\$8000	MD delay	=\$8000	MD mov16	=\$8000
MD incl6	=\$8000	keybd	=\$C000	kbstrobe	=\$C010	? VBL	=\$C019
? spkr	=\$C030	an0	=\$C058	an1	=\$C05A	dsend	=\$C05A
? an2	=\$C05C	? an3	=\$C05E	pb0	=\$C061	pb1	=\$C062
drecv	=\$C062	? pb2	=\$C063	? ptrig	=\$C070	dsk6off	=\$C0E8
zipslow	=\$C0E8	ERROR	=\$D412	FIXLINKS	=\$D4F2	RUNPROG	=\$D566
ADDON	=\$D998	FRMNUM	=\$DD67	SYNCHR	=\$DEC0	SYNERR	=\$DEC9
PTRGET	=\$DFE3	COLDSTRT	=\$E000	GETBYT	=\$E6F8	GETADR	=\$E752
SETFOR	=\$EB27	FLO2	=\$EBA0	? PRBL2	=\$F94A	? PREAD	=\$FB1E
? HOME	=\$FC58	CROUT1	=\$FD8B	PRBYTE	=\$FD8B	COUT	=\$FDED
? BELL	=\$FF3A						

